



Guía de Calc 26.2

Capítulo 9. Fórmulas y funciones



Derechos de Autor

Este documento tiene derechos de autor © 2026 por el equipo de documentación de LibreOffice. Los colaboradores se listan más abajo. Se puede distribuir y modificar bajo los términos de la GNU General Public License versión 3 o posterior o la Creative Commons Attribution License, versión 4.0 o posterior. Todas las marcas registradas mencionadas en esta guía pertenecen a sus propietarios legítimos.

Colaboradores

De esta edición

Olivier Hallot.

De esta edición (traducción y revisión)

B. Antonio Fernández.

De ediciones previas

Edward Olson, Vasudev Narayanan, Mike Kaganski, Barbara Duprey, Jean Hollis Weber, John A Smith, Olivier Hallot, Kees Kriek, Steve Fanning, Leo Moons, Gordon Bates, Felipe Viggiano, Rachel Kartch, Skip Masonsmith, B. Antonio Fernández.

De ediciones previas (traducción y revisión)

Jazmín Hurtado, Celia Palacios, Juan. Carlos Sanz Cabrero, Victor Cueto Jara, B. Antonio Fernández.

Comentarios y sugerencias

Puede dirigir cualquier comentario o sugerencia acerca de este documento al [foro del equipo de documentación en español](#) (es necesario registrarse).



Nota

Todo lo que publique en este foro, su dirección de correo o cualquier información personal escrita en el mensaje se archiva públicamente y no puede ser borrada.

Fecha de publicación y versión del programa

Versión en español publicada en febrero de 2026. Basada en la versión 26.2 de LibreOffice.

Uso de LibreOffice en macOS

Algunas pulsaciones de teclado y opciones de menú en macOS, son diferentes de las usadas en Windows y Linux. La siguiente tabla muestra algunas equivalencias para las instrucciones usadas en este capítulo. Para una lista detallada vea la ayuda de la aplicación.

<i>Windows o Linux</i>	<i>Equivalente en Mac</i>	<i>Efecto</i>
Herramientas > Opciones	LibreOffice > Preferencias	Acceso a la configuración general
<i>Clic</i> con el botón derecho	<i>Ctrl + clic</i> (o <i>clic derecho</i>)	Abre el menú contextual
<i>Ctrl</i> (Control)	 (Comando)	Utilizado con otras teclas
<i>F5</i>	<i>Mayús++F5</i>	Abre el navegador del documento
<i>F11</i>	 + <i>T</i>	Abre la <i>Barra lateral</i> en la página <i>Estilos</i>

CONTENIDO

Derechos de Autor.....	2
Introducción.....	5
Acerca de los Cálculos.....	5
Cálculo automático.....	6
Cómo funciona el cálculo automático.....	6
Caso de uso de cálculo automático.....	6
Procedimientos de cálculo automático.....	6
Problema de precisión inherente.....	7
Configurar una hoja de cálculo.....	7
Entradas en las fórmulas.....	7
Documentación.....	8
Fórmulas para el control de errores.....	8
Escribir fórmulas.....	9
Operadores en las fórmulas.....	9
Tipos de operadores.....	10
Operadores aritméticos.....	10
Operadores comparativos.....	11
Operadores de texto.....	11
Operadores de referencia.....	12
Operador de intervalos de celdas.....	12
Operador de concatenación de referencias.....	13
Operador de intersección.....	13
Ejemplos.....	13
Referencias relativas y absolutas.....	14
Referencia relativa.....	14
Referencia absoluta.....	15
Intervalos con nombre.....	16
Nombrar rangos rápidamente.....	18
Expresiones con nombres.....	19
Orden de cálculo.....	19
Cálculos que enlazan hojas.....	20
Comprender las funciones.....	23
Compatibilidad con otras aplicaciones de hojas de cálculo.....	24
Estructura de funciones.....	24
Funciones anidadas.....	25
Uso de la Lista de funciones de la Barra lateral.....	26
Uso del Asistente para funciones.....	26
Fórmulas matriciales.....	29
¿Qué es una fórmula matricial?.....	29
¿Cuándo usar fórmulas matriciales?.....	30
Crear fórmulas matriciales.....	30
Intersección implícita de fórmulas matriciales.....	30
Estrategias para la creación de fórmulas y funciones.....	31
Coloque una fórmula única en cada celda.....	32
Divida las fórmulas en partes y combine las partes.....	32
Optimizar cálculos.....	32
Utilice fórmulas matriciales para grandes conjuntos de datos.....	32
Utilice funciones de consolidación.....	32
Utilice las macros para crear funciones.....	33

Múltiples subprocesos.....	33
Encontrar y corregir errores.....	33
Mensajes de error.....	33
Errores habituales y posibles soluciones.....	34
#DIV/0! División por cero.....	34
#VALOR (sin resultado) y #REF: Referencias Incorrectas.....	35
Codificación de colores.....	36
Resaltado de valores.....	36
El Detective.....	36
Ejemplos de funciones.....	37
Aritmética y estadística básica.....	37
Aritmética básica.....	37
Estadísticas simples.....	38
Uso de estas funciones.....	39
Redondeo de cifras.....	39
Funciones volátiles y no volátiles.....	40
Uso de comodines y expresiones regulares en funciones.....	40
Funciones avanzadas.....	43

Introducción

En los capítulos anteriores trabajamos en cada celda con uno de los dos tipos básicos de datos: números y textos. Sin embargo, no siempre sabremos cuáles deberían ser los contenidos. A menudo, el contenido de una celda depende del contenido de otras celdas. Para manejar esta situación usamos un tercer tipo de datos: la fórmula.

Las fórmulas son ecuaciones que usan números y variables para obtener un resultado. En una hoja de cálculo, las variables son las referencias a celdas que contienen los datos necesarios para completar la ecuación.

Una función es un cálculo preestablecido introducido en una celda con el fin de analizar o manipular datos. Todo lo que resta por hacer es facilitar los argumentos y valores para que el cálculo se realice automáticamente. Las funciones le ayudan a crear las fórmulas necesarias para obtener los resultados que está buscando.

Acerca de los Cálculos

Calc calcula y recalcula datos a través de un proceso que involucra varios componentes y mecanismos clave. Aquí hay un desglose de cómo funciona en rasgos generales:

- 1) **Seguimiento de dependencia:** Cuando se crea una fórmula en una celda, Calc realiza un seguimiento de qué celdas son datos de entrada para esa fórmula. Esto se conoce como seguimiento de dependencia. Por ejemplo, si la celda A1 contiene una fórmula que suma valores de A2 y A3, la hoja de cálculo sabe que A1 depende de A2 y A3.
- 2) **Detección de cambios:** Cuando se modifican los datos de cualquier celda (ya sea manualmente o mediante otra fórmula), Calc detecta el cambio. Identifica qué celdas se ven afectadas por este cambio en función de las dependencias establecidas.
- 3) **Activación del recálculo:** Cuando se detecta un cambio, Calc activa un nuevo cálculo de todas las celdas afectadas. Dependiendo de la configuración y el contexto, puede realizar un cálculo completo (actualizando todas las fórmulas) o parcial (actualizando solo las celdas que dependen de las celdas modificadas).
- 4) **Orden de cálculo:** El cálculo se produce en un orden específico, normalmente siguiendo la cadena de dependencia. Primero se calcularán las celdas que no dependen de otras, seguidas de las celdas que dependen de ellas.
- 5) **Modos de recálculo:**
 - **Recálculo automático:** Este es el modo predeterminado en Calc. Los cambios en los valores de entrada activan automáticamente nuevos cálculos de las celdas dependientes. Compruébelo en **Datos > Calcular > Cálculo automático**.
 - **Recálculo Manual:** En este modo, puede controlar cuándo se vuelve a calcular. Después de realizar una serie de cambios, el usuario puede activar manualmente un recálculo. Vaya a **Datos > Calcular – Recálculo** o pulse **F9**. Para forzar el recálculo de todas las fórmulas en el documento, incluidas las funciones complementarias y no volátiles, use **Datos > Calcular – Recálculo incondicional** o pulse **Ctrl+Mayús+F9**.
- 6) **Visualización de resultados:** Después del recálculo, Calc actualiza los valores de las celdas respectivas según los datos y fórmulas actuales. Puede ver estos valores actualizados en tiempo real. Los resultados se formatean según los atributos de formato de celda, como lugares decimales, color de letra y formatos de números.
- 7) **Optimización del rendimiento:** Para conjuntos de datos grandes o cálculos complejos, Calc implementa varias técnicas de optimización para acelerar el recálculo.

Esto podría incluir el almacenamiento en caché de los resultados o el uso de algoritmos inteligentes para reducir la cantidad de cálculos necesarios. Por ejemplo, la biblioteca OpenCL (si está disponible en la computadora) o el algoritmo de suma de Kahan.

Cálculo automático

El cálculo automático es una función incorporada en LibreOffice Calc que garantiza que las fórmulas se actualicen instantáneamente cada vez que cambian los valores en las celdas dependientes. Esta funcionalidad mejora la precisión en tiempo real y reduce la necesidad de nuevos cálculos manuales.

Cómo funciona el cálculo automático

- Cuando un usuario modifica una celda, cualquier fórmula vinculada actualiza automáticamente sus valores.
- Calc evalúa continuamente las dependencias para garantizar que todas las celdas afectadas reflejen las últimas entradas.
- Este proceso es particularmente útil para conjuntos de datos dinámicos, como modelos financieros, seguimiento de inventario o paneles de control en vivo..

Caso de uso de cálculo automático

El impacto del cálculo automático varía según la complejidad de la hoja de cálculo.

¿Cuándo es beneficioso el cálculo automático?

- Modelos y paneles financieros: garantiza actualizaciones en tiempo real, lo que lo hace indispensable para hojas de cálculo dinámicas.
- Colaboración y libros de trabajo compartidos: evita datos obsoletos cuando varios usuarios modifican valores simultáneamente.
- Reducción de errores: Reduce el riesgo de errores de cálculo debido a valores no actualizados.

¿Cuándo es útil deshabilitar el cálculo automático?

- Grandes conjuntos de datos y fórmulas complejas: los recálculos continuos pueden ralentizar el rendimiento.
- Informes y resúmenes estáticos: cuando los datos no cambian con frecuencia, los recálculos manuales (F9) pueden optimizar los recursos.
- Gestión de recursos del sistema: ayuda a prevenir una carga excesiva de la CPU en modelos financieros complejos.

Equilibrar el cálculo automático para mayor eficiencia:


Para entornos altamente interactivos y basados en datos, el cálculo automático es valioso. Sin embargo, para informes y análisis de datos históricos, los recálculos manuales garantizan un rendimiento más fluido.

Procedimientos de cálculo automático

La función de cálculo automático se puede habilitar o deshabilitar a través de: **Datos > Calcular > Cálculo automático**. Alternativamente, al pulsar F9 se realiza un recálculo manual, lo que permite actualizaciones selectivas para un mejor control del rendimiento.

Un icono en la barra de estado indica si el *Cálculo automático* está desactivado (figura 1). Para habilitar la función, haga clic en el icono.

Figura 1: Icono de Cálculo automático en la barra de estado



Problema de precisión inherente

Calc, al igual que la mayoría de los programas de hojas de cálculo, utiliza capacidades matemáticas de punto flotante disponibles en el hardware. Dado que la mayoría del hardware contemporáneo utiliza aritmética binaria de punto flotante con precisión limitada definida en IEEE 754, muchos números decimales (incluidos tan simples como 0,1) no se pueden representar con precisión en Calc (que utiliza números de doble precisión de 64 bits internamente). Los cálculos con esos números necesariamente dan como resultado errores de redondeo, y esos se acumulan con cada cálculo.

Esto no es un error, pero se espera y actualmente es inevitable sin utilizar cálculos complejos en el software, lo que incurriría en penalizaciones de rendimiento inapropiadas y, por lo tanto, está fuera de discusión. Debe tener esto en cuenta y utilizar redondeos y comparaciones con ϵ según sea necesario.

Por ejemplo, la fórmula simple de la columna A:

A
31.000,99
32.000,12
=A1-A2

El resultado en A3 es -999.129999999997, en lugar del -999.13 esperado (es posible que deba aumentar los decimales mostrados en el formato de celda para ver esto).

Configurar una hoja de cálculo

Si está configurando algo más que una simple hoja de cálculo en Calc, vale la pena que lo planifique con anticipación, y que siga unas pautas primordiales:

- Elija el tipo correcto de entrada en las fórmulas.
- Incluya documentación.
- Incorpore un sistema de control de errores.

Entradas en las fórmulas

Muchos usuarios escriben largas y complejas fórmulas con valores fijos escritos directamente en la fórmula. Por ejemplo, la conversión de una moneda a otra requiere conocer la tasa de conversión vigente. Si introduce en la celda C1 la fórmula `= 0,75*B1` (para calcular el valor en euros de una cierta cantidad en dólares estadounidenses introducida en la celda B1), tendría que editar la fórmula cuando el tipo de cambio suba o baje de 0,75.

Es mucho más fácil configurar una celda donde especifique la tasa de cambio y que esta sea la celda de referencia en cualquier fórmula que necesite esa tasa. Los cálculos del tipo «qué pasaría si...» también se simplifican: ¿qué pasa si el tipo de cambio varía de 0,75 a 0,70 o a 0,80? No es necesario editar la fórmula y está claro cuál es la tasa que se debe usar en los cálculos.

Desglosar las fórmulas complejas en partes más manejables, también ayuda a minimizar los errores y ayuda a solucionar problemas.

Documentación

La falta de documentación es un error muy común. Muchos usuarios preparan una hoja de cálculo simple que luego se convierte en algo mucho más complejo con el tiempo. Sin documentación, el propósito y la metodología originales no estarán claros y pueden ser difíciles de descifrar. En este caso, generalmente es más fácil comenzar de nuevo, desde el principio, aunque se pierda el trabajo anterior.

Si inserta comentarios en las celdas o usa etiquetas y encabezados, puede hacer más fácil de comprender y modificar la hoja de cálculo (por usted o por terceras personas) y ahorrará mucho tiempo y esfuerzo en el futuro.

Fórmulas para el control de errores

Agregar columnas de datos o selecciones de celdas a partir de una hoja de trabajo a veces provoca errores debido a la omisión de celdas, la especificación incorrecta de un intervalo o un recuento doble de celdas. Por dicha razón es útil incluir procesos de verificación en sus hojas de cálculo.

Por ejemplo, prepare una hoja de cálculo para calcular la suma por columnas y use la función SUMA para calcular los totales parciales por columnas. Puede verificar el resultado al incluir (en una columna que no se imprima) un conjunto de totales de fila y sumarlos. Las dos cifras (total de fila y total de columna) deben coincidir. Si no lo hacen, hay un error en alguna parte.

Incluso puede configurar una fórmula para que calcule la diferencia entre los dos totales e informe un error en caso de que se devuelva un resultado distinto de cero (figura 2).

Figura 2: Control de errores en fórmulas

Demostración de comprobación de error				
Suma de columnas A B y C				
	A	B	C	Suma de filas
	0	0,64	0,02	0,66
	0,43	0,23	0,75	1,41
	0,91	0,57	0,59	2,07
	0,07	0,07	0,45	0,59
	0,37	0,33	0,04	0,74
	0,34	0,06	0,98	1,38
	0,95	0,34	0,65	1,94
	0,93	0,08	0,63	1,64
	0,61	0,82	0,17	1,6
Suma de columnas	=SUMA(B\$10:B\$17)	2,5	4,26	
		Total:	11,37	12,03
		Comprobación:	'!!!ERROR !!!	

Calc proporciona una ayuda para localizar problemas con las fórmulas en la hoja de cálculo, como un resalte visual de las celdas que contengan fórmulas. Esta opción está deshabilitada de forma predeterminada. Para habilitarla, vaya a **Herramientas > Opciones > LibreOffice Calc > Ver** y marque la opción *Indicador de fórmula y pista* en la sección *Mostrar*. Calc dibuja un triángulo azul en la esquina inferior izquierda de cada celda que contiene una fórmula. Al pasar el puntero sobre el triángulo, se muestra la fórmula en un cuadro emergente (incluso está seleccionada otra celda).

Figura 3: Indicador de fórmula

Radio		2,5
Circunferencia		15,707963267949
Área		19,6349540849362
		=PI()*POTENCIA(I1;2)

Escribir fórmulas

Puede escribir fórmulas de dos maneras:

- Utilizando el Asistente para funciones o la página Funciones de la Barra lateral
- Escribiendo directamente en la celda o en la línea de entrada.

Una fórmula debe comenzar con el símbolo igual (=). Sin embargo, en el caso de escribir directamente, también puede comenzar la fórmula con un más (+) o un menos (-). Por ejemplo, si ingresa `-2*A1`, Calc agregará automáticamente el símbolo igual a la fórmula.

Tenga presente que:

- Si solo ingresa un número positivo o negativo, como `-2` o `+3`, no se trata como fórmula
- Comenzar con un carácter distinto de `=`, `-` o `+` se tratará como texto, no como fórmula.

Operadores en las fórmulas

Cada celda de la hoja puede contener datos o instrucciones para el cálculo de datos. Para introducir datos, simplemente escriba en la celda y muévase a la siguiente celda o pulse la tecla *Intro*. Con las fórmulas, el signo igual (=) al comienzo de la celda indica que esta se usará para un cálculo. Un ejemplo de un cálculo matemático como `15+46` se muestra en la figura 4.

Figura 4: Cálculo simple (Columna B) y por referencias (columna D)

	A	B	C	D	
1					
2					
3		=15+46		15	
4				46	
5				=D3+D4	
6					

	A	B	C	D	
1					
2					
3		61		15	
4				46	
5				61	
6					

En la parte superior de la figura se muestran los datos y fórmulas introducidos y en la parte inferior el resultado final. Mientras que para el cálculo en la columna B se emplea solo una celda para la fórmula, la potencia de la hoja de cálculo se muestra en la columna D, donde los datos se colocan en dos celdas y el cálculo se realiza usando las referencias a las celdas; en este caso, las celdas D3 y D4 contienen los datos. Observe que en la celda D5, la fórmula es `= D3+D4` (el signo más indica que las celdas D3 y D4 se deben sumar).

El resultado aparecerá en la celda que contenga la fórmula. Todas las fórmulas se basan en este concepto. En la tabla 1 se muestran otras formas básicas de fórmulas.

Las referencias a celdas permiten que las fórmulas utilicen datos desde cualquier lugar de la hoja en la que se está trabajando o desde cualquier otra hoja del libro de cálculo abierto. Si los datos necesarios están en hojas de trabajo diferentes, se referencia el nombre de la hoja de trabajo, por ejemplo `=Hoja2.B12 + Hoja3.A11`.

Nota

Para que el signo (=) se muestre como texto y no como parte de una fórmula, escriba un apóstrofo o comilla simple, antes del signo igual. Por ejemplo, al escribir '= significa cosas diferentes para diferentes personas', Calc trata todo lo posterior al apóstrofo (incluido el signo =) como texto.

Tabla 1: Formas habituales de fórmulas

Fórmula	Descripción
= A1+10	Suma 10 al contenido de la celda A1.
= A1*16%	Calcula el 16% del contenido de la celda A1.
= A1*A2	Multiplica el contenido de la celda A1 por el de A2.
= REDONDEAR(A1; 1)	Redondea el contenido de la celda A1 con un decimal.
= INT.EFFECTIVO(5%; 12)	Calcula el interés efectivo del 5% de interés nominal anual con 12 pagos al año.
= B8-SUMA(B10:B14)	Resta al contenido de la celda B8 la suma de las celdas B10 a B14.
= SUMA(B8; SUMA(B10:B14))	Suma el contenido de las celdas B10 a B14 y luego agrega ese resultado al contenido de B8.
= SUMA(B1:B1048576) = SUMA(B:B)	Cualquiera de las dos fórmulas proporciona la suma todos los números en la columna B.
= MEDIA(MisDatos)	Muestra el promedio de un intervalo de celdas con el nombre <i>MisDatos</i> . Es posible establecer intervalos por inclusión nombrándolos en la entrada de menú Hojas>Intervalos y expresiones con nombres>Definir . – <i>MisDatos</i> representa un intervalo definido como por ejemplo B3:B10.
=SI(C31>140; "MAYOR"; "OK")	Las funciones lógicas como es el caso de SI ofrecen un resultado de un análisis condicional basado en los datos de la celda identificada. Si el contenido de C31 es mayor que 140, entonces se muestra MAYOR, de lo contrario se muestra OK.

Las funciones de la tabla 1 se identifican mediante una palabra, por ejemplo REDONDEAR, seguida de paréntesis que encierran los parámetros para la función separados por un separador de función que puede ser coma (,) o punto y coma (;) dependiendo de su configuración.

Tipos de operadores

En Calc se pueden usar los siguientes tipos de operadores:

- aritmético,
- comparativo,
- de texto
- de referencia.

Operadores aritméticos

Los operadores de suma, resta, multiplicación y división devuelven resultados numéricos. Los operadores de negatividad y porcentaje identifican una característica del número referenciado por la celda, por ejemplo -37. El ejemplo de potenciación ilustra cómo ingresar un número que se multiplica por sí mismo una cierta cantidad de veces. Por ejemplo, 2^3 o $2^3 = 2*2*2$.

Tabla 2: Operadores aritméticos.

Operador	Nombre	Ejemplo
+ (más)	Adición	=1+1
- (menos)	Sustracción	=2-1
- (menos)	Negatividad	-5
* (asterisco)	Multiplicación	=2*2
/ (barra)	División	=10/5
% (porcentaje)	Porcentaje	15%
^ (acento circunflejo)	Potenciación	2^3

Operadores comparativos

Los operadores comparativos se encuentran en fórmulas que usan la función SI y devuelven un valor en función del resultado de la comparación (verdadero o falso); por ejemplo, =SI(B6>G12, 127, 0) significa que si el contenido de la celda B6 es mayor que el contenido de la celda G12, devolverá el número 127 y en caso contrario el número 0.

Se puede obtener una respuesta directa de VERDADERO o FALSO ingresando una fórmula como =B6>B12. Si los números que se encuentran en las celdas referenciadas están representados con precisión, la respuesta será VERDADERO; de lo contrario, será FALSO.

Tabla 3: Operadores Comparativos

Operador	Nombre	Ejemplo	Resultado (A=4, B=5)
=	Igual	A1 = B1	FALSO
>	Mayor que	A1 > B1	FALSO
<	Menor que	A1 < B1	VERDADERO
>=	Mayor o igual que	A1 >= B1	FALSO
<=	Menor o igual que	A1 <= B1	VERDADERO
<>	Desigual	A1 <> B1	VERDADERO

Si la celda A4 contiene el valor numérico 4 y la celda B1 contiene el valor 5, las operaciones anteriores arrojarían resultados de FALSO, FALSO, VERDADERO, FALSO, VERDADERO y VERDADERO.

Operadores de texto

Es común que los usuarios coloquen texto en las hojas de cálculo. Para ofrecer otras opciones de qué y cómo se muestra este tipo de datos, el texto se puede unir en piezas procedentes de diferentes lugares de la hoja de cálculo. La figura 5 muestra un ejemplo.

Figura 5: Concatenación de texto

F2								
	A	B	C	D	E	F	G	H
1								
2		23	junio	2020		=B2 & " de " & C2 & " de " & D2		
3								
F2								
	A	B	C	D	E	F	G	H
1								
2		23	junio	2020		23 de junio de 2020		
3								

En este ejemplo, se encontraron partes específicas del texto en tres celdas diferentes. Para unir estos segmentos, la fórmula = B2 & " de " & C2 & " de " & D2 utiliza el operador & y también agrega los espacios y la puntuación (de) necesarios entre comillas, lo que da resultado deseado. El efecto es, la concatenación de una fecha con un formato particular.

Calc dispone también de la función **CONCATENAR** que realiza la misma operación.

Operadores de referencia

Una celda se identifica por el identificador de columna (letra) ubicado en los encabezados de columnas y un identificador de fila (número) situado en los encabezados de fila. En hojas de cálculo leídas de izquierda a derecha, la referencia para la celda superior izquierda es **A1**.

En su forma más simple, una referencia puede referirse tanto a una única celda como a un intervalo de celdas (selección rectangular ej. **A1:B5**), un intervalo 3D (referencias con varias hojas) o a una referencia obtenida de una lista de intervalos no contiguos. Para construir estas referencias complejas, necesita operadores de referencia.

Operador de intervalos de celdas

El operador de intervalos de celdas se escribe con el signo dos puntos (:). Una expresión que utiliza el operador de intervalo de celdas tiene la sintaxis:

referencia superior izquierda : referencia inferior derecha.

El operador de intervalo de celdas crea una referencia al intervalo más pequeño que incluye las celdas comprendidas entre la celda de referencia izquierda y la celda de referencia derecha.

En la esquina superior izquierda de la figura 6 se muestra la referencia **A1:D10**, correspondiente a las celdas incluidas en la operación de arrastre con el ratón para resaltar el intervalo.

Figura 6: Operador de referencia para un intervalo de celdas




Tabla 4: Ejemplos de operador de intervalos

Ejemplo	Descripción
A2:B4	Referencia a un intervalo de celdas rectangular con 6 celdas, 2 columnas de ancho × 3 filas de alto. Cuando hace clic en la referencia en la fórmula en la línea de entrada, un borde indica el rectángulo referenciado.
(A2:B4):C9	Referencia a un intervalo de celdas rectangular con la celda A2 arriba a la izquierda y la celda C9 abajo a la derecha. El intervalo de celdas contiene 24 celdas, 3 columnas de ancho × 8 filas de alto. Este método de direccionamiento amplía el intervalo de celdas inicial de A2:B4 a A2:C9.
Hoja1.A3 : Hoja3.D4	Referencia a un intervalo de celdas cuboide con 24 celdas, 4 columnas de ancho × 2 filas de altura × 3 hojas de profundidad. (Suponiendo que las hojas Hoja1, Hoja2 y Hoja3 aparezcan en ese orden)
B:B	Referencia a todas las celdas de la columna B.

Ejemplo	Descripción
A:D	Referencia a todas las celdas de la columna A a la columna D.
20:20	Referencia a todas las celdas de la fila 20.
1:20	Referencia a todas las celdas de las filas 1 a 20.

Cuando ingrese B4:A2, B2:A4 o A4:B2 directamente, Calc lo cambiará a A2:B4 para que la celda superior izquierda del intervalo de celdas esté a la izquierda del signo dos puntos (:) y la celda inferior derecha quede a su derecha. Pero si nombra la celda B4, por ejemplo, como *_inicio*, con el símbolo guión bajo (_) al inicio del nombre y nombra a la celda A2 como *_final*, con el símbolo guión bajo (_) al final del nombre, puede usar *_inicio : _final* sin ningún error. Para más información, vea el título «Intervalos con nombre» más adelante.

Operador de concatenación de referencias

El operador de concatenación es el símbolo tilde o virgulilla (~). Una expresión que utiliza el operador de concatenación (~) tiene la siguiente sintaxis:

referencia izquierda ~ referencia derecha

El resultado de dicha expresión es una lista ordenada de referencias. Algunas funciones pueden tomar una lista de referencias como argumento, por ejemplo SUMA, MAX o INDICE.

La concatenación de referencias a veces se llama *unión*. Pero no es la unión de los dos conjuntos «referencia izquierda» y «referencia derecha», como se entiende en la teoría de conjuntos. La fórmula = CONTAR(A1:C3 ~ B2:D2) devuelve 12, (9 + 3 celdas), pero en la unión de dos conjuntos equivaldría a 10 celdas.

Observe que la fórmula = SUMA(A1:C3 ; B2:D2) es diferente de = SUMA(A1:C3 ~ B2:D2) aunque dan el mismo resultado. La primera es una llamada a función con 2 parámetros y cada uno de ellos hace referencia a un intervalo de celdas. La segunda es una llamada a función con 1 parámetro, que es una lista de referencias.

La concatenación de referencias también se aplica a filas y columnas completas. Por ejemplo, = SUMA(A:B ~ D:D) es la suma de todas las celdas de las columnas A y B y la columna D.

Operador de intersección

El operador de intersección se escribe como un signo de cierre de exclamación (!). Una expresión que utiliza el operador de intersección (!) tiene la siguiente sintaxis:

referencia izquierda ! referencia derecha

Si las referencias apuntan a intervalos únicos, el resultado es una referencia a un solo intervalo de celdas que contiene todas las celdas que están tanto en la referencia izquierda como en la referencia derecha.

Si las referencias son listas de referencias, entonces cada elemento de la lista de la izquierda se cruza con cada uno de la derecha y estos resultados se concatenan a una lista de referencias. El orden es cruzar primero el primer elemento de la izquierda con todos los elementos de la derecha, luego cruzar el segundo elemento de la izquierda con todos los elementos de la derecha y así sucesivamente.

Ejemplos

A2:B4 ! B3:D6

Da como resultado una referencia al intervalo de celdas B3: B4, porque estas celdas están dentro de A2: B4 y dentro de B3: D6. Esto se puede apreciar en la figura 7, en la que las celdas del intervalo A2:B4 tienen fondo naranja y las celdas del intervalo

B3:D6 tienen bordes negros. Las celdas que poseen tanto fondo naranja como el borde negro (B3:B4) forman la intersección de los dos intervalos.

Figura 7: Ejemplo de operador de intersección

	A	B	C	D	E
1					
2					
3					
4					
5					
6					
7					

(A2:B4 ~ B1:C2) ! (B2:C6 ~ C1:D3)

Primero se calculan las intersecciones A2:B4!B2:C6, A2:B4!C1:D3, B1:C2!B2:C6 y B1:C2!C1:D3. Esto da como resultado B2:B4, vacío, B2:C2 y C1:C2. Posteriormente, estos resultados se concatenan, omitiendo las partes vacías. De esta forma, el resultado final es la lista de referencia es B2:B4 ~ B2:C2 ~ C1:C2.

A:B ! 10:10

Calcula la intersección de las columnas A y B con la fila 10, seleccionando así A10 y B10.

Puede utilizar el operador de intersección para hacer referencia a una celda en una tabulación cruzada de una manera comprensible. Por ejemplo, si tiene columnas etiquetadas como «Temperatura y Precipitación» y las filas etiquetadas como «enero, febrero, marzo», etc. La siguiente expresión

'febrero' ! 'Temperatura' hará referencia a la celda que contiene la temperatura en febrero. la expresión es equivalente a 3:3 ! B:B (siendo 3 la fila correspondiente a febrero y B la columna correspondiente a las temperaturas)

El operador de intersección (!) tiene mayor precedencia que el operador de concatenación (~), pero no depende de la precedencia.



Consejo

Escriba siempre entre el paréntesis la parte que se va a calcular primero.

Referencias relativas y absolutas

Las referencias son la forma en que conocemos la dirección o ubicación de una celda en Calc. Pueden ser relativas (a su posición en relación con otras celdas) o absolutas (a su posición fija).

Referencia relativa

Un ejemplo de una referencia relativa ilustrará la diferencia entre una referencia relativa y una referencia absoluta usando la hoja de cálculo de la figura 8.

- 1) Escriba los números 4 y 11 en las celdas C2 y C3 respectivas de esa hoja de cálculo.
- 2) Copie la fórmula de la celda B4 (= B2+B3) en la celda C5. Puede hacerlo usando un simple copiar y pegar o haciendo clic en B4 y arrastrando el pequeño punto cuadrado inferior derecho hasta C4. La fórmula en B4 calcula la suma de valores en las celdas B2 y B3.
- 3) Haga clic en la celda C4. Verá que la barra de fórmulas muestra = C2+C3 en lugar de = B2+B3 y el valor en C4 es 15 (la suma de C2 y C3).

En la celda B4, las referencias a las celdas B2 y B3 son referencias relativas. Esto significa que Calc interpreta la fórmula en B4 como la suma de las dos celdas inmediatamente superiores en esa columna (B). Cuando copia la fórmula en la celda C4, Calc usa el mismo procedimiento: calcula la suma de los valores de las celdas superiores (C2 y C3). Esta vez, el resultado es 15.

Figura 8: Referencia relativa

	A	B	C	D	E
1					
2		15	4		
3		46	11		
4		61			
5					

	A	B	C	D	E
1					
2		15	4		
3		46	11		
4		61	15		
5					
6					

Piense que una fórmula con direcciones relativas utiliza unidades de distancia desde la celda donde está la fórmula. La celda B1 está a 1 columna a la izquierda de la celda C4 y 3 filas arriba; esta dirección podría describirse como R[-3]C[-1] (filas = -3 y columnas = -1). De hecho, este método de notación se utilizaba anteriormente en las hojas de cálculo.

Siempre que copie la fórmula de la celda B4 a otra celda, el resultado siempre será la suma de los dos números tomados de las dos celdas situadas una y dos filas arriba respecto a la celda que contiene la fórmula.

Las referencias relativas es el método predeterminado para referencias de direcciones en Calc.

Referencia absoluta

Es posible que desee multiplicar una columna de números por una cantidad fija señalada en una sola celda de la hoja de cálculo. Por ejemplo, una columna de cifras puede mostrar cantidades en dólares estadounidenses. Para convertir estas cantidades a euros es necesario multiplicar cada monto en dólares por el tipo de cambio. Es decir, 10,00 \$ se multiplicaría por 0,75 para convertir a euros, en este caso 7,50 €. El siguiente ejemplo muestra cómo ingresar una tasa de cambio y usar esa tasa para convertir cantidades en una columna, de dólares (USD) a euros.

- 1) Ingrese 0,75 como el tipo de cambio euros a dólares USD en la celda D1. Ingrese los montos (en USD) en las celdas D2, D3 y D4, por ejemplo, 10, 20 y 30.
- 2) En la celda E2, escriba la fórmula = D2*D1. El resultado que se muestra correctamente es 7,5.

Figura 9: Ingrese la fórmula de conversión adecuada en E2 y cópiela en E3.

	C	D	E	F	G
1		0,75 €			
2		\$10,00	7,50 €		
3		\$20,00			
4		\$30,00			

- 3) Copie la fórmula de la celda E2 en la celda E3. El resultado que aparece es 200, lo que, por supuesto, es incorrecto. Calc ha copiado la fórmula usando referencias relativas: la fórmula en E3 es = D3*D2 y esto no es lo que queremos, lo que queremos es multiplicar D3 por D1, y por tanto, que se multiplique cada celda de la columna de dólares un mismo factor, la celda D1.

Figura 10: Resultado en E3 incorrecto; cambie la fórmula en E2 a una referencia absoluta

E3			$f_x \Sigma \nabla =$	=D3*D2	SI			$f_x \times \checkmark$	=D2*\$D\$1
	C	D	E	F		C	D	E	F
1		0,75 €				1	0,75 €		
2		\$10,00	7,50 €			2	\$10,00	=D2*\$D\$1	
3		\$20,00	200,00 €			3	\$20,00	200,00 €	
4		\$30,00				4	\$30,00		

- 4) Edite la fórmula de la celda E2, para que sea = D2*\$D\$1. Cópielo en las celdas E3 y E4. Los resultados son 15 y 22.5, lo correcto.

Figura 11: Resultado correcto con referencias absolutas

E2					E3				
		C	D	E	F	G			H
1			0,75 €				1		0,75 €
2			\$10,00	7,50 €			2		\$10,00 7,50 €
3			\$20,00	200,00 €			3		\$20,00 15,00 €
4			\$30,00				4		\$30,00

Los signos \$ antes de la D y del 1 convierten la referencia a la celda D1, antes relativa, a absoluta o fija. Si la fórmula se copia en otra celda, la segunda parte siempre mostrará \$D\$1. La interpretación de esta fórmula para las celdas E2, E3 y E4 es que toma el valor de la celda de la misma fila, una columna a la izquierda (primer factor de multiplicación) y lo multiplica por el valor de la celda D1.

Las referencias a celdas pueden tener cuatro formas, como se muestra en la siguiente tabla:

Tabla 5: Tipos de referencia de celda

Referencia	Tipo de direccionamiento
D1	Relativo: desde E3, la celda una columna a la izquierda y dos filas arriba.
\$D\$1	Absoluto: es la celda D1.
\$D1	Parcialmente absoluto: desde E3, la celda de la columna D y dos filas arriba.
D\$1	Parcialmente absoluto: desde E3, la celda una columna a la izquierda y la fila 1.

Consejo

Para cambiar las referencias en las fórmulas, resalte las celdas y pulse **F4** para alternar las cuatro formas de referencia. Para alternar solo una parte de la fórmula seleccione la celdas en la barra de fórmulas y utilice **F4**. La entrada del menú **Hoja > cambiar entre tipos de referencia de celda** es el equivalente a pulsar **F4**

El conocimiento de las diferencias entre referencias relativas y absolutas es fundamental si desea copiar y pegar fórmulas o vincular hojas de cálculo.

Intervalos con nombre

Las celdas y los intervalos de celdas pueden tener un nombre asignado. Nombrar celdas e intervalos mejora la legibilidad de la fórmula y el mantenimiento de los documentos.

Un ejemplo simple sería nombrar el intervalo de celdas B1:B10 como *Peso* para sumar todas las celdas con pesos. La fórmula inicial es = SUMA(B1:B10). Cuando se ha nombrado el intervalo de celdas B1:B10 como *Peso*, la fórmula se puede transformar en = SUMA(Peso). La legibilidad de las fórmulas resulta más clara.

Otra ventaja es que todas las fórmulas que tienen el intervalo de celdas con nombre como argumento se actualizan cuando el intervalo de celdas con nombre cambia de ubicación o

tamaño. Por ejemplo, si el intervalo de celdas *Peso* se cambia a las celdas **P10:P30**, no es necesario que revise todas las fórmulas que tienen *Peso* como argumento; solo necesita modificar el intervalo de celdas nombrado *Peso* con el nuevo tamaño o ubicación.

Para definir una celda o intervalo de celdas con nombre:

- 1) Seleccione la celda o intervalo de celdas
- 2) Vaya al menú **Hoja > Intervalos y expresiones con nombre > Definir**
- 3) En el diálogo que aparece (figura 12), especifique el nombre y alcance.

Figura 12: Diálogo Definir nombre

También puede definir un intervalo de celdas con nombre directamente en la hoja, seleccionando el intervalo de celdas y a continuación escribiendo el nombre en el cuadro de nombres de la barra de fórmulas (figura 13).

Figura 13: Nombrar un intervalo desde la barra de fórmulas

	A	B	C
1		20	
2		21	
3		22	
4		23	
5		24	
6		25	
7		26	
8		27	
9		28	
10		29	
11			

Para un acceso rápido a un intervalo de celdas con nombre, seleccione el nombre de su interés en el cuadro de nombres, a la izquierda de la barra de fórmulas de la hoja. El intervalo de celdas nombrado se muestra en la pantalla y se selecciona.

Para modificar un intervalo con nombre, utilice el diálogo *Gestionar nombres* (figura 14). Se accede a este diálogo en el menú **Hoja > Intervalos y expresiones con nombre > Gestionar**, con la flecha del cuadro nombres (barra de fórmulas) o con **Ctrl+F3**.

Figura 14: Diálogo Gestionar nombres

Nombre	Intervalo o expresión de fórmula	Alcance
Peso	\$Hoja2.\$B\$1:\$B\$10	Documento (Global)

Seleccione celdas en el documento para actualizar el intervalo.

Nombre:

Intervalo o expresión de fórmula:

Alcance:

☒ Opciones del intervalo

Nombrar rangos rápidamente

Si su tabla de datos incluye encabezados de columna en la parte superior o inferior y encabezados de fila a la izquierda o a la derecha, puede crear intervalos con nombres rápidamente usando el diálogo *Crear nombres*:

1. Seleccione el intervalo e incluya los encabezados de la tabla.
2. Vaya a **Hoja > Intervalos y expresiones con nombre > Crear** en el menú. Se abrirá el diálogo *Crear Nombres*(figura 15).
3. Especifique la ubicación de los encabezados. Los intervalos con nombre se crearán automáticamente.

Nota

Si los encabezados contienen nombres duplicados, Calc le pedirá que confirme si desea reemplazar las apariciones anteriores con la más reciente.

Figura 15: Diálogo Crear nombres

Crear nombres a partir de

☒ Fila superior

☐ Columna izquierda

☐ Fila inferior

☐ Columna derecha

Consejo

Si la tabla incluye encabezados de fila y columna, puede hacer referencia a cualquier elemento utilizando el operador de intersección de rango (!), por ejemplo: `=encabezado_fila ! encabezado_columna`.

Expresiones con nombres

También puede usar un nombre para una fórmula larga o compleja. Para nombrar una fórmula, abra el diálogo *Definir nombre* (figura 12) e ingrese la expresión de la fórmula en el cuadro de texto *Intervalo o expresión de fórmula*. Nombre la expresión y haga clic en *Añadir*.

Como ejemplo, suponga que necesita calcular en las celdas C1 a C10 la circunferencia de un conjunto de círculos y tiene los diferentes radios en el área B1 a B10. Defina una expresión con nombre *CIRCUNFERENCIA*, con de fórmula = 2*PI() *B1 y haga clic en *Añadir* para cerrar el diálogo. En la celda C1, escriba = CIRCUNFERENCIA y pulse la tecla *Intro*. La fórmula se aplica a la celda C1. Copie la celda y péguela en las celdas restantes de C2 a C10: tendrá las circunferencias de todos los círculos. Todas las celdas en el intervalo de celdas C1:C10 tienen la expresión de fórmula = CIRCUNFERENCIA.

Nota

Tenga en cuenta que la expresión nombrada de fórmula usa las mismas reglas para el direccionamiento de celdas, es decir, referencias absolutas o relativas.

Orden de cálculo

El *orden de cálculo* se refiere a la secuencia en la que se realizan las operaciones en una fórmula. Para más información al respecto, lea el siguiente artículo de Wikipedia: https://es.wikipedia.org/wiki/Orden_de_evaluación.

La división y la multiplicación se realizan antes que la suma o la resta. Existe una tendencia bastante habitual a esperar que los cálculos se realicen de izquierda a derecha, ya que la ecuación se leería en español. Calc evalúa la fórmula completa; luego, basándose en la precedencia de operaciones, descompone la fórmula ejecutando las operaciones de multiplicación y división antes que las otras.

Por lo tanto, al crear fórmulas, debe probar su fórmula para asegurarse de que se obtiene el resultado correcto. Es una buena práctica probarlo a fondo. Si es necesario, utilice paréntesis para aclarar el orden de las operaciones y hacer explícita su intención. La siguiente tabla demuestra cómo se aplica el orden de cálculo en la práctica.

Tabla 6 – Orden de cálculo

Intención de cálculo (de izquierda a derecha)	Cálculo siguiendo el orden de Calc
1+3*2+3 = 11 se pretende: 1+3 = 4, luego 4*2 = 8, luego 8+3 = 11.	= 1+3*2+3, resultado 10. Es decir: 3*2 = 6, luego 1+6+3 = 10.
Otra posible intención podría ser: 1+3*2+3 = 20 Es decir: 1+3 = 4, luego 2+3 = 5, luego 4*5 = 20	El programa resuelve la multiplicación de 3*2 antes de resolver la suma de los números.

Si desea que el resultado sea una de las dos posibles soluciones de la izquierda, ordene la fórmula como sigue:

$$((1+3) * 2)+3 = 11$$

$$(1+3) * (2+3) = 20$$

Nota

Use paréntesis para agrupar las operaciones en el orden de cálculo que pretenda; por ejemplo, = B4+G12*C4/M12 puede definirse mejor como =((B4+G12)*C4)/M12.

Cálculos que enlazan hojas

Otra característica poderosa de Calc es la capacidad de vincular datos a través de varias hojas de trabajo. El nombrar las hojas puede ser útil para identificar dónde se pueden encontrar datos específicos. Un nombre como *Nómina* o *Venta de Plantas* es mucho más significativo que *Hoja1*. La función llamada **HOJA** devuelve el número de hoja (la posición) en el conjunto de hojas del libro de cálculo. Puede haber varias hojas en cada libro y se numeran desde la izquierda: Hoja1, Hoja2, etc. Si arrastra las hojas de trabajo a diferentes ubicaciones entre las pestañas, la función devuelve el número que se refiere a la posición de esta hoja en el libro de cálculo. En una nueva instancia de Calc, el valor predeterminado es una sola hoja de trabajo.

Por ejemplo, si escribe `= HOJA()` en A1 en la Hoja1, la celda devuelve el valor **1**. Si arrastra la Hoja1 para colocarla entre las hojas 2 y 3, el valor cambia a **2**; que es segunda hoja del libro.

Se puede encontrar un ejemplo con cálculos que obtienen datos de otras hojas en un entorno empresarial donde una empresa combina ingresos y costos de cada una de las operaciones de sus sucursales en una hoja de trabajo combinada. Observe las cuatro figuras siguientes:

Figura 16: Hoja que contiene datos de la sucursal 1

	A	K	L	M	N
1	Creciendo Verde S. L.				
2	Ventas hasta la fecha				
3		oct	nov	dic	Hasta la fecha
4	Ingresos:				
5	Venta de plantas	36.288	52.874	81.335	1.283.107
6	Venta de fertilizantes	16.822	3.825	3.600	697.634
7	Venta de tierra	2.019	459	432	84.479
8	Total parcial	55.129	57.158	85.367	2.065.220
9					
10	Coste de ventas				
11	Compras al mayorista	18.744	19.434	29.025	702.175
12	Impuestos ventas	6.064	6.287	9.390	227.174
13	Total parcial	24.808	25.721	38.415	929.349
14					
15	Total ingresos:	30.321	31.437	46.952	1.135.871
16					
17	Gastos:				

Figura 17: Hoja que contiene datos de la sucursal 2

	A	K	L	M	N
1	Creciendo Verde S. L.				
2	Ventas hasta la fecha				
3		oct	nov	dic	Hasta la fecha
4	Ingresos:				
5	Venta de plantas	38.251	14.899	49.588	1.027.538
6	Venta de fertilizantes	6.120	2.384	7.934	164.406
7	Venta de tierra	734	286	952	19.729
8	Total parcial	45.105	17.569	58.474	1.211.673
9					
10	Coste de ventas				
11	Compras al mayorista	15.336	5.973	19.881	411.969
12	Impuestos ventas	4.962	1.933	6.432	133.284
13	Total parcial	20.298	7.906	26.313	545.253
14					
15	Total ingresos:	24.807	9.663	32.161	666.420
16					
17	Gastos:				

Figura 18: Hoja que contiene datos de la sucursal 3

	A	K	L	M	N
1	Creciendo Verde S. L.				
2	Ventas hasta la fecha				
3		oct	nov	dic	Hasta la fecha
4	Ingresos:				
5	Venta de plantas	65801	58.257	102.179	1.498.444
6	Venta de fertilizantes	54833	17.620	8.782	843.175
7	Venta de tierra	59025	16.824	7.622	397.342
8	Total parcial	179.659	92.701	118.583	2.738.961
9					
10	Coste de ventas				
11	Compras al mayorista	61.084	31.518	40.318	931.246
12	Impuestos ventas	19.762	10.197	13.044	301.285
13	Total parcial	80.846	41.715	53.362	1.232.531
14					
15	Total ingresos:	98.813	50.986	65.221	1.506.430
16					
17	Gastos:				

Figura 19: Hoja con los datos combinados de todas las sucursales


K5					
	A	K	L	M	N
1	Creciendo Verde S. L.				
2	Ventas hasta la fecha				
3		oct	nov	dic	Hasta la fecha
4	Ingresos:				
5	Venta de plantas	140.340	126.030	233.102	3.809.089
6	Venta de fertilizantes	77.775	23.829	20.316	1.705.215
7	Venta de tierra	61.778	17.569	9.006	501.550
8	Total parcial	279.893	167.428	262.424	6.015.854
9					
10	Coste de ventas				
11	Compras al mayorista	95.164	56.925	89.224	2.045.390
12	Impuestos ventas	30.788	18.417	28.866	661.743
13	Total parcial	125.952	75.342	118.090	2.707.133
14					
15	Total ingresos:	153.941	92.086	144.334	3.308.721
16					
17	Gastos:				

Las hojas se han montado con estructuras idénticas. La forma más sencilla de hacer esto es abrir una nueva hoja de cálculo, configurar la primera hoja de la sucursal Hoja1, ingresar los datos, formatear las celdas y preparar las fórmulas para las distintas sumas de filas y columnas.

A continuación, cree copias a partir de la primera hoja del siguiente modo:

- 1) En la pestaña de la hoja de trabajo, haga clic con el botón derecho y seleccione *Cambiar nombre de hoja*. Escriba *Sucursal1*. Haga clic derecho en la pestaña nuevamente y seleccione *Mover o Copiar hoja*.
- 2) En el diálogo *Mover/copiar hoja* (figura 20), seleccione la opción *Copiar* (se selecciona automáticamente si solo hay una hoja) y en el área *Insertar antes* seleccione *desplazar a la última posición*. Cambie la entrada en *Nombre nuevo* a *Sucursal2* y pulse el botón *Aceptar*. Repita los pasos anteriores para las hojas *Sucursal3* y *Combinadas*.

Figura 20: Copiar una hoja en el mismo libro




- 3) Ingrese los datos de la sucursal 2 y de la sucursal 3 en las hojas respectivas. Cada hoja es independiente e informa los resultados de las sucursales individuales.
- 4) En la hoja *Combinadas*, haga clic en la celda K5. Escriba el signo =, haga clic en la pestaña *Sucursal1*, haga clic en la celda K5, pulse el signo +. Repita para las hojas *Sucursal2* y *Sucursal3* y pulse *Intro*. Ahora tiene una fórmula en la celda K5 que agrega los ingresos de las ventas de plantas para las tres sucursales (figura 21)

Figura 21: Vinculación entre los ingresos de octubre de las hojas de sucursal

K5		= \$'Sucursal 1'.K5+\$'Sucursal 2'.K5+\$'Sucursal 3'.K5				
	A	K	L	M	N	O
1	Creciendo Verde S. L.					
2	Ventas hasta la fecha					
3		oct	nov	dic	Hasta la fecha	
4	Ingresos:					
5	Venta de plantas	140.340	52.874	81.335	1.283.107	
6	Venta de fertilizantes	16.822	3.825	3.600	697.634	
7	Venta de tierra	2.019	459	432	84.479	
8	Total parcial	159.181	57.158	85.367	2.065.220	
9						
10	Coste de ventas					
11	Compras al mayorista	18.744	19.434	29.025	702.175	
12	Impuestos ventas	6.064	6.287	9.390	227.174	
13	Total parcial	24.808	25.721	38.415	929.349	
14						
15	Total ingresos:	134.373	31.437	46.952	1.135.871	
16						
17	Gastos:					

- 5) Copie la fórmula de K5 y resalte el intervalo de celdas K5:N15. Use el atajo del teclado **Ctrl+Alt+Mayús+V** o haga clic en **Editar > Pegado especial**. Asegúrese de que la casilla *Fórmulas* esté marcada y desmarque las casillas *Todo* y *Formatos* del diálogo *Pegado especial*. Pulse el botón *Aceptar*. A continuación verá el mensaje de advertencia de sobrescritura (figura 22):

Figura 22: Pegar una fórmula en un intervalo de celdas



- 6) Haga clic en el botón *Sí*. Ahora ha copiado las fórmulas en cada celda mientras mantiene el formato configurado (figura 23). Por supuesto, en este ejemplo tendría que ordenar la hoja de trabajo eliminando los ceros en las filas sin formato.

Figura 23: Pegado especial desde la celda K5:N15

N15						
	A	K	L	M	N	O
1	Creciendo Verde S. L.					
2	Ventas hasta la fecha					
3		oct	nov	dic	Hasta la fecha	
4	Ingresos:					
5	Venta de plantas	140.340	126.030	233.102	3.809.089	
6	Venta de fertilizantes	77.775	23.829	20.316	1.705.215	
7	Venta de tierra	61.778	17.569	9.006	501.550	
8	Total parcial	279.893	167.428	262.424	6.015.854	
9		0	0	0	0	
10	Coste de ventas	0	0	0	0	
11	Compras al mayorista	95.164	56.925	89.224	2.045.390	
12	Impuestos ventas	30.788	18.417	28.866	661.743	
13	Total parcial	125.952	75.342	118.090	2.707.133	
14		0	0	0	0	
15	Total ingresos:	153.941	92.086	144.334	3.308.721	
16						
17	Gastos:					

También se puede utilizar el *Asistente para funciones* para realizar la vinculación. El uso del asistente se describe en detalle en el título «Uso del Asistente para funciones».

Comprender las funciones

Calc incluye más de 500 funciones para ayudarle a analizar y referenciar datos. Muchas de estas funciones se utilizan con números, pero otras se utilizan con fechas y horas o incluso con texto. Una función puede ser tan simple como sumar dos números o encontrar el promedio de una serie de números o puede ser tan compleja como calcular la desviación estándar de una muestra o la tangente hiperbólica de un número.

Normalmente, el nombre de una función es una descripción abreviada de lo que hace la función. Por ejemplo, la función **VF** da el valor futuro de una inversión, mientras que **BIN.A.HEX** convierte un número binario en un número hexadecimal. Por tradición, las funciones se ingresan completamente en mayúsculas, aunque Calc las leerá correctamente si también están en minúsculas o en una mezcla de mayúsculas y minúsculas.

Algunas funciones básicas son algo similares a los operadores. Ejemplos:

- +** (más) El operador suma dos o más datos para obtener el resultado. La función **SUMA** también suma grupos de números en intervalos contiguos.
- *** (por) El operador multiplica dos o más datos para obtener el resultado. La función **PRODUCTO** actúa igual que la función **SUMA** pero multiplicando.

Cada función usa varios argumentos para realizar los cálculos. Estos argumentos pueden tener o no su propio nombre. A usted le corresponde ingresar los argumentos necesarios para ejecutar correctamente la función.

En algunos casos, los argumentos tienen opciones predefinidas y es posible que necesite revisar el texto del *Asistente para funciones* o consultar la Ayuda del programa para entenderlos. Sin embargo, a menudo un argumento es un valor que se ingresa manualmente o uno que ya ingresó en una celda o intervalo de celdas. En Calc, puede ingresar valores de otras celdas escribiendo su nombre o intervalo de celdas o en algunos casos, seleccionando celdas con el ratón. Si los valores de las celdas cambian, el resultado de la función se actualiza automáticamente.

Consejo

Puede encontrar más detalles sobre cada una de las funciones disponibles en el área Funciones de Calc de The Document Foundation Wiki en https://wiki.documentfoundation.org/Documentation/Calc_Functions/es.

Compatibilidad con otras aplicaciones de hojas de cálculo

El conjunto de funciones Calc sigue el estándar *OpenFormula* (Fórmula Abierta) definido en la sección 2 «[Recalculated Formula \(OpenFormula\) Forma](#)» del formato de documento abierto (ODF) versión 1.2, Accesible en las publicaciones de las páginas de [OASIS](#) o de las normas ISO <https://www.iso.org/standard/66375.html>.

El apoyo general de Calc a OpenFormula conduce a un nivel de compatibilidad inherente con el conjunto de funciones de cualquier otra aplicación de hoja de cálculo que siga el mismo estándar. (Hay algunas funciones dentro de Calc que no están de acuerdo con OpenFormula, pero muchas de ellas se incluyen específicamente para mejorar el intercambio de archivos entre Calc y Microsoft Excel, que no sigue esta norma.

A pesar de las diferencias entre aplicaciones, Calc puede abrir hojas de cálculo creadas por muchas aplicaciones diferentes y puede guardar hojas de cálculo en distintos formatos.

Por ejemplo, cuando Calc abre un archivo de Excel que utiliza la función `MULTIPLO.SUPERIOR` de Excel, estas se convierten automáticamente en la función de referencia de Calc `MULTIPLO.SUPERIOR.XCL`. De manera similar cuando Calc guarda una hoja de cálculo que usa la función `MULTIPLO.INFERIOR`, ya que estas se convierten automáticamente en la función de referencia de Excel. `MULTIPLO.INFERIOR.MAT`.

La wiki de The Document Foundation proporciona una página comparando las características de LibreOffice y Microsoft Office: https://wiki.documentfoundation.org/Feature_Comparison:_LibreOffice_-_Microsoft_Office. Esta comparación muestra que Calc proporciona actualmente 508 funciones discretas, de las cuales solo 30 son exclusivas de Calc y el resto tienen sus contrapartes en Microsoft Excel. Es evidente que existe un alto nivel de coincidencia entre los conjuntos de funciones de Calc y Excel y que muchas funciones se pueden utilizar en ambas aplicaciones sin ningún cambio, aumentando así la compatibilidad.

En algunos casos en los que una función de Calc produce un resultado de acuerdo con las normas internacionales, pero difiere del realizado por la función equivalente en Excel. En tales situaciones Calc suele tener una función similar pero con un modificador adecuado añadido al nombre (como «`_ADD`» o «`EXCEL2003`»), las cuales proporcionan el mismo resultado en Excel.

Estructura de funciones

Todas las funciones tienen una estructura similar. Si usa la herramienta adecuada para ingresar una función, puede evitar memorizar la estructura, pero es mejor conocerla para solucionar posibles problemas.

Como ejemplo típico, la estructura de la función que puede encontrar celdas que coinciden con los criterios de búsqueda es de la siguiente manera:

= `BDCONTAR(Base de datos;Campo de base de datos;Criterios de búsqueda)`

Puntos clave sobre la estructura de funciones:

- **Las Funciones son parte de fórmulas:** una función no puede existir por sí sola; siempre debe estar en una fórmula. Incluso si la función es la fórmula completa, debe haber un signo igual (=) al comienzo antes de la función.
- **Nombre de función:** la función comienza con su nombre, como `BDCONTAR` en el ejemplo anterior.

- **Argumentos:** después del nombre de la función vienen las entradas necesarias (argumentos), los argumentos son obligatorios, se agregan entre paréntesis y están separados por punto y coma o coma según su configuración.
- Una función de Calc puede tener hasta 255 argumentos y Los argumentos pueden ser:
 - números
 - una celda
 - Intervalos con nombre, intervalos o matrices (que pueden contener varias o incluso cientos de celdas).

El tipo y formato de los argumentos dependen de la naturaleza de la función, la siguiente tabla muestra cómo se ingresan los argumentos en función de las necesidades de la función:

Tabla 7: Entrada de los argumentos de una función

Argumento	Descripción
"texto"	Las comillas indican que se está ingresando una cadena de texto.
9	El número nueve se ingresa como número.
"9"	El número nueve se ingresa como texto.
A1	Se está ingresando la dirección de la celda A1 y el argumento de la función tomará lo que contenga la celda A1.
B2:D9	Se ha ingresado un intervalo rectangular de celdas.
Ventas_2	Un intervalo con nombre previamente definido

Funciones anidadas

Las funciones también se pueden utilizar como argumentos dentro de otras funciones. Cuando esto ocurre se denominan funciones anidadas.

= SUMA(2;PRODUCTO(5;7))

Para tener una idea de lo que pueden hacer las funciones anidadas, imagine que está diseñando un módulo de aprendizaje autodidacta. Durante el módulo, los estudiantes hacen tres pruebas e ingresan los resultados en las celdas A1, A2 y A3. En A4, puede crear una fórmula anidada que comienza promediando los resultados de las pruebas: = PROMEDIO(A1:A3). Luego, la fórmula usa la función SI para dar al estudiante retroalimentación que depende de la calificación promedio en las pruebas. La fórmula completa diría:

= SI(PROMEDIO(A1:A3) > 85;"¡Felicitaciones! Está listo para avanzar al siguiente módulo."; "Falló. Revise el material nuevamente. Si es necesario, comuníquese con su instructor para obtener ayuda.")

Dependiendo del promedio obtenido, el estudiante recibiría el mensaje de felicitación o de error.

Observe que la fórmula anidada para el promedio no requiere de otro signo igual. El que está al comienzo de la ecuación es suficiente para ambas funciones.

Si es nuevo en las hojas de cálculo, la mejor manera de pensar en las funciones es como un lenguaje de programación. Hemos utilizado ejemplos sencillos para explicar el concepto con más claridad, pero con las funciones anidadas, una fórmula de Calc puede volverse bastante compleja.




Nota

Calc mantiene la sintaxis de la fórmula como un mensaje emergente cerca de la celda donde está escribiéndola, como una ayuda nemotécnica mientras la escribe.

Uso de la Lista de funciones de la Barra lateral

Un método confiable para ingresar las fórmulas es mediante la página *Lista de funciones* de la *Barra lateral* (figura 24). Puede acceder a ella desde el menú **Vista > Lista de funciones** o haciendo clic en el icono *Funciones*.

Figura 24: Lista de funciones en la Barra lateral de Calc



La *Lista de funciones* incluye una breve descripción de cada función y sus argumentos. Resalte la función y mire en la parte inferior para ver la descripción. Si es necesario, coloque el cursor sobre la división entre la lista y la descripción; cuando el cursor se convierta en una flecha de dos puntas, arrástrelo hacia arriba para aumentar el espacio de la descripción. Haga doble clic en el nombre de una función para agregarla a la celda actual, junto con los marcadores de posición para cada uno de los argumentos de la función.

Calc proporciona muchas funciones y la página *Funciones* de la *Barra lateral* proporciona una función de búsqueda. Cuando escribe en el cuadro de búsqueda en la parte superior, la lista de funciones se reduce a cada carácter insertado.

El uso de la *Lista de funciones* es casi tan rápido como la entrada manual y tiene la ventaja de que no es necesario que memorice una fórmula que desee utilizar. En teoría, también debería ser menos propenso a errores. En la práctica, sin embargo, algunos usuarios pueden tener problemas al reemplazar los marcadores de posición con valores.

Otra característica de la *Lista de funciones* es la capacidad de mostrar las últimas fórmulas utilizadas para un acceso más rápido.

Uso del Asistente para funciones

El método de entrada más utilizado para ingresar funciones en las fórmulas es el *Asistente para funciones* (figura 25). Para abrirlo puede usar alguno de los siguientes tres métodos: vaya a **Insertar > Función** o haga clic en el botón *Asistente para funciones* en la barra de fórmulas o pulse **Ctrl+F2**.


El *Asistente para funciones* proporciona las mismas características de ayuda que la *Lista de funciones*, pero agrega campos en los que puede ver el resultado de la función ya completada, así como el resultado de fórmulas anidadas.

- 1) Seleccione una categoría de funciones o escriba en *Buscar* para acortar la lista

- 2) Desplácese hacia abajo por las funciones nombradas y seleccione la requerida haciendo doble clic en ella.

Cuando selecciona una función, la descripción aparece en el lado derecho del diálogo. De forma opcional, puede escribir el nombre de la función en el cuadro de búsqueda: esta se reducirá por cada carácter insertado (figura 25).

Figura 25: Página Funciones en el Asistente para funciones




En la tabla 8 se muestran las categorías disponibles del Asistente para funciones y el número de funciones disponibles en cada una de ellas.

Tabla 8: Categoría de funciones en el Asistente para funciones

Categoría	funciones	Categoría	funciones
Complemento	48	Lógica	11
Matriz	15	Matemáticas	82
Base de datos	12	Hoja de cálculo	22
Fecha y hora	36	Estadísticas	151
Finanzas	63	Texto	47
Información	21		

El asistente muestra un área a la derecha donde puede ingresar datos manualmente en cuadros de texto o puede hacer clic en el botón *Reducir* (figura 26), situado a la derecha de los argumentos, para cambiar el asistente a su modo de selección de celdas en la hoja de trabajo.

Figura 26: Asistente para funciones en modo selección



- 3) Para seleccionar celdas, haga clic en el campo directamente en la celda o mantenga pulsado el botón izquierdo del ratón y arrastre para seleccionar el área requerida. Al soltar el botón del ratón, aparecerá la referencia del intervalo seleccionado.
- 4) Después de seleccionar el área, haga clic en el botón *Reducir* para volver al asistente.
- 5) Si se necesitan ingresar varios argumentos, haga clic en el siguiente cuadro de texto y repita el proceso de selección para la siguiente celda o intervalo. Repita este proceso tantas veces como sea necesario. El asistente aceptará hasta 255 intervalos como argumentos en la función **SUMA**.
- 6) Finalmente, pulse el botón *Aceptar* para aceptar la función y agregarla a la celda y obtener el resultado.

Nota

Si selecciona una función desde la lista haciendo doble clic en ella y luego selecciona una diferente haciendo doble clic nuevamente, la segunda fórmula de opción se agrega a la fórmula de primera opción en el cuadro de texto *Fórmula*. Si no desea una función anidada, debe borrar lo ingresado anteriormente en el cuadro de fórmula y luego hacer doble clic en la nueva función.

Esta característica aditiva le permite crear fórmulas con funciones anidadas de una manera más fácil y visual en el cuadro de texto *Fórmula*.

Dentro del *Asistente de funciones* también puede seleccionar la pestaña *Estructura* para una vista en árbol de las partes de la fórmula. La principal ventaja sobre la *Lista de funciones* es que cada argumento se ingresa en su propio campo, lo que facilita su administración. Aunque la entrada sea más lenta, al crear una hoja de cálculo, la precisión es generalmente más importante que la velocidad.

La vista de estructura del asistente es importante para depurar y corregir fórmulas largas, anidadas y complejas. En esta vista, la fórmula se analiza y cada componente de la fórmula se calcula mediante una llamada de función más simple u operación aritmética y luego se combina siguiendo las reglas de cálculo. Se puede visualizar cada elemento de la fórmula que se analiza y comprobar si los resultados intermedios son correctos, hasta encontrar el error.

Las funciones también se pueden ingresar directamente en la *Línea de entrada*. Después de ingresarla pulse la tecla *Intro* o haga clic en el botón *Aplicar* situado en la barra de fórmulas, para agregar la función a la celda y obtener el resultado (figura 27).

Figura 27: Barra de fórmulas



1 2 3 4 5

- (1) Cuadro de nombre: muestra la lista de funciones comunes
- (2) Asistente para funciones

- (3) Cancelar
- (4) Aplicar
- (5) Línea de entrada

Si en lugar del resultado ve las fórmulas en las celdas, deshabilite la casilla *Fórmulas* en **Herramientas > Opciones > LibreOffice Calc > Ver > Mostrar**. Esta opción de mostrar las fórmulas es útil para auditar la hoja de cálculo o solucionar errores. Aunque deshabilite la casilla *Fórmulas* y se muestre el resultado en la celda, aún puede ver la fórmula en la línea de entrada.

Sugerencia

La opción del menú **Ver > Mostrar fórmula** o mediante **Ctrl+`** (acento grave), también sirve para alternar la visibilidad de las fórmulas en las celdas de la hoja.

Fórmulas matriciales

¿Qué es una fórmula matricial?

Una fórmula en la que se evalúan los valores individuales de un intervalo de celdas se denomina *fórmula matricial*. La diferencia entre una fórmula matricial y otras fórmulas es que la primera trata con varios valores simultáneamente en lugar de solo uno. Esto significa que al menos uno de los argumentos debe ser una matriz, es decir, un intervalo de celdas sobre el que operará simultáneamente.

Además, una fórmula matricial no solo puede procesar varios valores, sino que también puede devolver varios valores. El resultado de una fórmula matricial también es una matriz.

Cuando Calc actualiza las fórmulas, se lee cada celda afectada y su fórmula se vuelve a calcular. Si tiene mil celdas en una columna con la misma fórmula (la expresión de la fórmula solo cambia los datos para calcular), termina con mil fórmulas idénticas para interpretar y ejecutar.

Las fórmulas matriciales evaluarán la fórmula una vez y ejecutarán los cálculos tantas veces como sea el tamaño de la matriz, ahorrando así el tiempo utilizado para interpretar cada fórmula de cada celda. Y debido a que Calc almacena solo una fórmula para todos los datos de la matriz de celdas, también ahorra espacio en el archivo del libro.

Para multiplicar por **10** cada uno de los valores de las celdas individuales de la matriz del lado izquierdo de la figura 28 no es necesario aplicar una fórmula a cada celda o valor individual. En su lugar, solo necesita usar una única fórmula matricial. Para ello seleccione un intervalo de celdas de 3x5 celdas en otra parte de la hoja de cálculo, ingrese la fórmula = **10*A1:C5** y confirme esta entrada usando **Ctrl+Mayus+Intro**, en lugar de solo **Intro**. El resultado es una matriz de 3x5 en la que cada valor del intervalo de celdas (A1:C5) se multiplica por **10**.

Figura 28: Matriz de origen resaltada en amarillo y la matriz resultante en verde. La fórmula matricial se visualiza en la barra de fórmulas

E1								
	A	B	C	D	E	F	G	
1	1	2	3		10	20	30	
2	2	3	4		20	30	40	
3	3	4	5		30	40	50	
4	4	5	6		40	50	60	
5	5	6	7		50	60	70	

Además de la multiplicación, también puede utilizar otros operadores en el intervalo (también llamado *matriz*) de celdas de referencia. Con Calc, puede sumar (+), restar (-), multiplicar (*), dividir (/), usar exponentes (^), concatenar (&) y hacer comparaciones (=, <>, <, >, <=, >=). Los operadores se pueden usar en cada uno de los valores individuales del intervalo; y devolverán el resultado como una matriz si se ingresó como fórmula matricial, es decir, si se finalizó la escritura de la fórmula con **Ctrl+Mayus+Intro** y no solo con la tecla **Intro**.

Los operadores de comparación en una fórmula matricial tratan a las celdas vacías de la misma manera que una fórmula normal, es decir, como cero o como una cadena vacía. Por ejemplo, si las celdas A1 y A2 están vacías, las fórmulas de matriz **{= A1:A2 = ""}** y **{= A1:A2 = 0}** devolverán una matriz de celdas de 1 columna y 2 filas que contienen textualmente **VERDADERO**.

¿Cuándo usar fórmulas matriciales?

Utilice fórmulas matriciales si tiene que repetir los mismos cálculos con valores diferentes. Si más adelante decide cambiar el método de cálculo, solo tiene que actualizar la fórmula matricial. Para agregar o modificar una fórmula matricial, seleccione todo el intervalo de celdas de la matriz y luego realice el cambio necesario en la fórmula matricial.

Las matrices son una herramienta esencial para realizar cálculos complejos, ya que puede tener diferentes intervalos incluidos en sus cálculos. Calc tiene diferentes funciones matemáticas para matrices, como la función `MMULT` que multiplica dos matrices.

Crear fórmulas matriciales

Si crea una fórmula matricial utilizando el *Asistente para funciones*, debe marcar la casilla de verificación *Matriz*, que aparece en la esquina inferior izquierda (figura 25). De lo contrario, la fórmula solo devolverá el valor de la celda superior izquierda de la matriz, que es donde se escribe el cálculo.

Si ingresa la fórmula de matriz directamente en la celda, se debe usar la combinación de teclas *Ctrl+Mayus+Intro* en lugar de la tecla *Intro*. Solo entonces la fórmula se convierte en una fórmula matricial.

 Nota

Las fórmulas matriciales aparecen entre llaves (**{ }**) en Calc de manera automática. No puede crear fórmulas matriciales ingresando manualmente las llaves.

 Nota

Las celdas de una matriz de resultados quedan protegidas automáticamente contra cambios. Sin embargo, puede editar o copiar la fórmula matricial en la barra de fórmulas si selecciona todo el intervalo de celdas de la matriz.

Intersección implícita de fórmulas matriciales

También puede crear una fórmula «normal» que use un intervalo de referencia, como parámetro, sin ser una fórmula matricial. Esta fórmula también se conoce como intersección implícita de una fórmula matricial.

El resultado se obtiene a partir de la intersección del intervalo y las filas o columnas en las que se encuentra la fórmula.

Si no hay intersección o si el intervalo en la intersección abarca varias filas o columnas, aparecerá un mensaje de error #¡VALOR!. El siguiente ejemplo ilustrará mejor lo expuesto :

Figura 29: Fórmula matricial implícita

E4			=A1:A3 + 100			
	A	B	C	D	E	
1	7	31	33	107		
2	95	17	2	195	195	
3	5	10	50	105		
4					#VALOR!	
5						

- 1) En D1, Inserte la siguiente fórmula = A1:A3 + 100 y pulse *Ctrl+Mayús+Intro* para hacer la fórmula matricial.
- 2) Las celdas D1, D2, D3 obtendrán los valores 107, 195 y 105, respectivamente porque se ha usado una fórmula matricial.
- 3) Escriba la fórmula = A1:A3 + 100 en E2, no lo ingrese como fórmula matricial.

- En la celda E4 aparecerá el mensaje #VALOR!, puesto que la fila 4 está fuera del intervalo de la usado en fórmula A1:A3.

Las fórmulas que hacen más que un simple cálculo o la suma de filas o columnas de valores generalmente requieren varios argumentos. Por ejemplo, considere la siguiente ecuación:

$$x = x_i + v_i t + \frac{1}{2} a t^2$$

Para facilitar la presentación, es una buena práctica configurar una hoja de cálculo de una manera similar a la que se muestra en la figura 29. En este ejemplo, las variables individuales se ingresan en las celdas de la hoja y no se requiere editar la fórmula (en la celda F6).

Puede adoptar varios enfoques generales al crear una fórmula. Al decidir qué enfoque tomar, considere cuántas personas necesitarán usar las hojas de trabajo, la duración de las hojas de trabajo y las variaciones que podrían encontrarse en el uso de la fórmula.

Si otras personas van a usar la hoja de cálculo, asegúrese de que sea fácil ver qué entrada se requiere y dónde. La explicación del propósito de la hoja de cálculo, la base del cálculo, la entrada requerida y la salida generada a menudo se colocan en la primera hoja de trabajo.

Figura 30: Configuración de una fórmula con argumentos

F6	$= \$C\$9 + \$C\$8 * E6 + 0,5 * \$C\$7 * E6^2$				
	A	B	C	D	E
1	Posición como una función de tiempo (t) para un movimiento lineal de traslación con aceleración constante				
2			$x = (\frac{1}{2}) * a * t^2 + v_i * t + x_i$		
3					
4	En verde: valores introducidos para el cálculo				
5	t	tiempo	variable independiente		
6	x	posición en el tiempo t	variable dependiente		
7	a	aceleración	16		
8	v_i	velocidad inicial	5		
9	x_i	posición inicial	50		
10					
11					
12					

Una hoja de cálculo que construya hoy, con muchas fórmulas complicadas, puede no ser tan obvia en su función y operación 6 o 12 meses después de ser haber sido creada. Utilice generosamente comentarios y notas para documentar su trabajo.

Es posible que sepa que no puede usar valores negativos o valores cero para un argumento en particular, pero si alguien más ingresa ese valor, es probable que la fórmula devuelva un mensaje de error. Una buena idea es utilizar una prueba lógica o un formato condicional para interceptar errores.

Coloque una fórmula única en cada celda

La estrategia más básica es considerar las fórmulas necesarias como simples y con una vida útil limitada. Para ello, coloque una fórmula única en cada celda apropiada. Esto es recomendable solo para hojas de cálculo muy simples o «desechables».

Divida las fórmulas en partes y combine las partes

La segunda estrategia es similar a la primera, usar fórmulas simples, pero dividiendo fórmulas largas en partes más pequeñas y luego combinarlas en el todo. Existen muchos ejemplos de este tipo en cálculos científicos y de ingeniería complejos donde los resultados provisionales se utilizan en varios lugares de la hoja. El resultado de calcular la velocidad de flujo del agua en una tubería se puede utilizar para estimar pérdidas debidas a la fricción, ya sea que la tubería fluya llena o parcialmente vacía y para optimizar el diámetro para el régimen de flujo dado.

En todos los casos, debe adoptar los principios básicos de creación de fórmulas descritos anteriormente.

Optimizar cálculos

Las hojas de cálculo se utilizan a menudo para procesar datos brutos y producir resúmenes, consolidar o visualizar información que después usará el tomador de decisiones o le servirá como fuente de informes. Los datos brutos pueden conseguirse mediante mediciones físicas, transacciones comerciales o recopilación de datos de diversa índole. Los cálculos llevados a cabo con estos datos pueden llevar mucho tiempo y durar minutos, horas y quizás días.

Un error común es insertar fórmulas para cada celda y realizar miles de interpretaciones y cálculos de fórmulas. A continuación se ofrecen algunas recomendaciones para acelerar los cálculos.

Utilice fórmulas matriciales para grandes conjuntos de datos

Las fórmulas matriciales tienen una fórmula aplicada a la masa de datos. El ahorro de cálculos puede ser significativo para grandes conjuntos de datos.

Utilice funciones de consolidación

Las funciones de consolidación realizan cálculos sobre conjuntos de datos. `SUMA`, `SUMAR.SI`, `SUMA.PRODUCTO` son ejemplos de funciones de consolidación. Por ejemplo, si tiene una lista de materiales muy larga, donde la cantidad debe multiplicarse por el precio unitario y luego totalizarse para producir una cifra de costo, en su lugar puede aplicar una fórmula en cada entrada de la lista de materiales y luego sumar con la fórmula `SUMA.PRODUCTO(cantidad ; precio unitario)`, donde la cantidad y el precio unitario serán intervalos de celdas que representan la lista de materiales. La función `SUMAPRODUCTO` multiplica cada celda del conjunto de datos de cantidad por su celda correspondiente de precio unitario y suma todos los productos.

Una situación similar ocurre cuando debe sumar un subconjunto del conjunto de datos original, donde debe aplicar una prueba en cada entrada para permitir que sea parte de la suma, por ejemplo, si el valor es estrictamente positivo. Utilice `SUMAR.SI(dato_a_probar;"> 0"; dato_a_sumar)`, donde `dato_a_probar` es el conjunto de datos donde se prueban los valores positivos, `dato_a_sumar` es la columna donde los valores se suman dependiendo de la prueba y `"> 0"` es la prueba en sí.

Otras funciones de consolidación son `SUMAR.SI.CONJUNTO`, `PROMEDIO.SI`, `CONTAR.SI`, `MIN.SI.CONJUNTO`, `MAX.SI.CONJUNTO` y otras.

Utilice las macros para crear funciones

Crear sus propias funciones y macros puede ser una estrategia poderosa para optimizar una hoja de cálculo y simplificar fórmulas. Este enfoque se debe utilizar cuando el resultado simplifique el uso de la hoja de cálculo por parte del usuario final y reduzca la probabilidad de errores. Además, esta configuración puede facilitar el mantenimiento al situar las correcciones o actualizaciones en una ubicación central.

El peligro del uso excesivo de macros o funciones personalizadas es que los principios en los que se basa la hoja de cálculo se vuelven mucho más complicados de comprender por un usuario que no sea el autor original y en algunas ocasiones incluso por el mismo autor.

El uso de macros es un tema especializado y se describe someramente en el «Capítulo 14».

Múltiples subprocesos

Muchos equipos modernos tienen procesadores de varios núcleos y proporcionan múltiples subprocesos. Los subprocesos múltiples hacen que las CPU sean más eficientes a la hora de brindar un mejor rendimiento general. Calc admite subprocesos múltiples para aprovechar cualquier procesamiento paralelo disponible en su equipo mientras trabaja con hojas de cálculo.

Esta función está controlada por la opción *Activar cálculo con subprocesos múltiples* en la sección de *Opciones de subprocesos de CPU* dentro del menú **Herramientas > Opciones > LibreOffice Calc > Calcular**. Está habilitado como valor predeterminado y se recomienda deshabilitarlo. Este es el único control en la interfaz de usuario de Calc que se relaciona con subprocesos múltiples; una vez iniciado, el procesamiento funciona automáticamente.

Si el cálculo con subprocesos múltiples está habilitado, Calc identificará automáticamente dónde podría aprovechar la hoja de cálculo los subprocesos múltiples y actuará consecuentemente. Los subprocesos se usan generalmente para grupos de fórmulas, donde suficientes celdas adyacentes en una columna usan la misma fórmula pero obtienen resultados diferentes debido al direccionamiento relativo de celdas. Una implicación de este enfoque es que la optimización se basa en columnas; un diseño basado en filas podría ser menos eficiente.

Hay otras formas de controlar la capacidad de subprocesos múltiples de Calc, como ajustar la variable de entorno específica MAX_CONCURRENCY de LibreOffice. Sin embargo, estos métodos están fuera del alcance de este documento.

Encontrar y corregir errores

Es común encontrar situaciones en las que se generan errores. Incluso con todas las herramientas disponibles en Calc para la ayuda en las fórmulas, es fácil cometer errores. A muchas personas les resulta difícil ingresar números y pueden cometer un error en el tipo de datos que precisa el argumento de una función. Además de corregir errores, se pueden buscar las celdas utilizadas en una fórmula para cambiar sus valores o verificar el resultado.

Calc proporciona tres herramientas para investigar las fórmulas y celdas referenciadas:

- 1) **Mensajes de error** que proporcionan información sobre problemas en las fórmulas..
- 2) **Codificación de colores** que resalta las celdas referenciadas para mejor identificación
- 3) **La herramienta Detective** que Ayuda a rastrear relaciones entre celdas e identificar errores.

Mensajes de error

La herramienta más básica son los mensajes de error. Los mensajes de error se muestran en la celda de una fórmula o en el *Asistente para funciones* en lugar del resultado.

Un mensaje de error para una fórmula suele ser un número de tres dígitos del 501 al 527 o, a veces, un fragmento de texto inútil como #¿NOMBRE?, #REF o #¡VALOR!. El número de error aparece en la celda y una breve explicación del error en el lado derecho de la barra de estado.

La mayoría de los mensajes de error indican un problema con la forma en que se ingresó la fórmula, aunque varios indican que se ha encontrado con una limitación de Calc o de su configuración actual.

Los mensajes de error no son fáciles de manejar y pueden intimidar a los nuevos usuarios. Sin embargo, son pistas valiosas para corregir errores. Puede encontrar explicaciones detalladas de ellos en la Ayuda en línea de LibreOffice, buscando *Códigos de error en Calc*. Algunos de los más comunes se muestran en la siguiente tabla.

Tabla 9: Mensajes de error más comunes

Código	Significado
#¿NOMBRE?	Se muestra en lugar de <i>Err: 525</i> . No existen referencias válidas para el argumento.
#REF	Se muestra en lugar de <i>Err: 524</i> . Falta la columna, fila u hoja de la celda a la que se hace referencia.
#¡VALOR!	Se muestra en lugar de <i>Err: 519</i> . El valor de uno de los argumentos no es el tipo que se requiere. Es posible que el valor se haya ingresado incorrectamente; por ejemplo, pueden faltar comillas alrededor del valor. En otras ocasiones, una celda o intervalo de celdas puede tener un formato incorrecto, como texto en lugar de números.
#¡DIV/0!	Se muestra en lugar de <i>Err: 532</i> . División por cero.
#¡NUM!	Un cálculo da como resultado un desbordamiento del intervalo de celdas de valores definido.
509	Falta un operador como un signo igual en la fórmula.
510	Falta una variable en la fórmula.

Errores habituales y posibles soluciones

#DIV/0! División por cero

Este error es el resultado de dividir un número entre cero (0) o usar una celda en blanco como denominador. Existe una forma sencilla de evitar este tipo de problemas. Cuando se muestra una celda en blanco o con valor cero, use una función condicional. En la figura 31 se muestra la división de las celdas de la columna B por las de la columna C, lo que produce 2 errores que surgen de un cero y una celda en blanco en la columna C.

Figura 31: Ejemplos de #DIV/0!: Error de división por cero

	A	B	C	D
1				
2	Fecha	Pacientes	Equipo enfermería	Pacientes por enfermera
3	01/05/22	24	5	4,8
4	02/05/22	16	5	3,2
5	03/05/22	21	3	7
6	04/05/22	17	0	#¡DIV/0!
7	05/05/22	18	4	4,5
8	06/05/22	17		#¡DIV/0!
9	07/05/22	19	5	3,8
10	08/05/22	22	4	5,5
11	09/05/22	21	4	5,25

Un error muy común, como este, surge de una situación en la que los datos no se han pasado a la función o se han pasado incorrectamente. Para evitar la aparición del error e informar sobre la falta de un dato, se puede usar la siguiente fórmula con la función condicional SI:

= SI(C3>0;B3/C3;"Sin informe")

La fórmula se copia sobre las celdas que pueden verse afectadas (columna D). El resultado de esta fórmula se ve en la figura 32 y su explicación es: si C3 es mayor que 0, B3 se divide entre C3; de lo contrario, se muestra el mensaje «Sin informe».

También es posible utilizar dos juegos de comillas como último parámetro para que aparezca un espacio en blanco (sin valor) o una fórmula diferente para sustituirlo por una respuesta estándar como el menor valor que pueda tener.

Figura 32: Solución al problema de la división por cero

D3				=SI(C3 > 0 ; B3 / C3 ; "Sin informe")
	A	B	C	D
1				
2	Fecha	Pacientes	Equipo enfermería	Pacientes por enfermera
3	01/05/22	24	5	4,8
4	02/05/22	16	5	3,2
5	03/05/22	21	3	7
6	04/05/22	17	0	Sin informe
7	05/05/22	18	4	4,5
8	06/05/22	17		Sin informe
9	07/05/22	19	5	3,8
10	08/05/22	22	4	5,5
11	09/05/22	21	4	5,25
12	10/05/22	18	3	6
13	11/05/22	19	4	4,75

#VALOR (sin resultado) y #REF: Referencias Incorrectas

El error #VALOR también es muy común. Un error de este tipo surge cuando una celda contiene un tipo de valor incorrecto. En el ejemplo de la figura 34, se ingresó el texto «Ninguno» en C9, y la fórmula de la columna D necesita que ese dato sea numérico.

Figura 33: Una entrada incorrecta causa el error #VALOR

	A	B	C	D
1				
2	Fecha	Pacientes	Equipo enfermería	Pacientes por enfermera
3	01/05/22	24	5	4,8
4	02/05/22	16	5	3,2
5	03/05/22	21	3	7
6	04/05/22	17	0	Sin informe
7	05/05/22	18	4	4,5
8	06/05/22	17	Ninguno	#¡VALOR!
9	07/05/22	19	5	3,8
10	08/05/22	22	4	5,5
11	09/05/22	21	4	5,25
12	10/05/22	18	3	6
13	11/05/22	19	4	4,75

El error #REF se debe a que falta una referencia. En el ejemplo de la figura 34, la fórmula hace referencia a una hoja que se ha eliminado.

Figura 34: Eliminar una hoja puede causar el error #REF!

SUMA				fx				=SUMA(\$'Hoja1'.A\$1:A\$10)				
A				B				C				
1												
2	=SUMA(\$'Hoja1'.A\$1:A\$10)											
3												

A2				fx				=SUMA(\$#;REF!.A\$1:A\$10)				
A				B				C				
1												
2	#REF!											
3												

Codificación de colores

Otra herramienta útil al revisar una fórmula en la *Línea de entrada* es la codificación de colores. Cuando selecciona una fórmula que ya ha sido ingresada, las celdas o intervalos usados para cada argumento en la fórmula se resaltan con el color seleccionado en la configuración.

Figura 35: Codificación de colores para las fórmulas

SI		fx		=SI(C4 > 0 ; B4 / C4 ; "Sin informe")		
	A	B	C	D	E	F
1						
2	Fecha	Pacientes	Equipo enfermería	Pacientes por enfermera		
3	01/05/22	24	5	4,8		
4	02/05/22	16	5	=SI(C4 > 0 ; B4 / C4 ; "Sin informe")		
5	03/05/22	21	2	7		

Calc utiliza ocho colores para delinear las celdas referenciadas, comenzando con azul para la primera celda y continuando con rojo, magenta, verde, azul oscuro, marrón, morado y amarillo antes de volver a recorrer la secuencia.

Resaltado de valores

Hay situaciones en las que la visualización del contenido de la celda es la misma aunque el tipo de datos sea diferente. Por ejemplo, un contenido de texto y un contenido numérico pueden tener el mismo aspecto, pero pueden producir errores si se utilizan en ciertos cálculos. Para ilustrar esto, considere una cadena de texto «10,35» alineada a la derecha en una celda se puede confundir con el valor numérico 10,35. Cuando se utiliza la celda en una fórmula, la cadena de texto se interpreta como valor cero y se produce un error.

El resaltado de valores distinguirá el texto y los tipos de datos numéricos asignando un color diferente a los caracteres del contenido. Para activarlo, vaya a **Ver > Destacar valores** en el menú o pulse **Ctrl+F8**. De forma predeterminada, el los caracteres de texto se muestran en negro, el contenido numérico en azul y los resultados de las fórmulas en verde. Consulte el «Capítulo 2, Introducir y editar datos» para más información sobre el resaltado de valores.

El Detective

En una hoja de cálculo extensa o complicada, la codificación de colores se vuelve menos útil. En estos casos, considere usar el *Detective* que se puede encontrar en el menú en **Herramientas > Detective**. El detective es una herramienta para verificar qué celdas se utilizan como argumentos en una fórmula (precedentes) y en qué fórmulas están anidados (dependientes). También se puede usar para monitorizar y rastrear errores, marcar datos no válidos (es decir, información en celdas que no está en el formato adecuado) o incluso para eliminar precedentes o dependientes.

Para usar esta herramienta, seleccione una celda con una fórmula, luego inicie el *Detective*. En la hoja de cálculo, verá líneas con sus terminaciones en círculos para indicar precedentes o en flechas para dependientes. Las líneas muestran el flujo de información.

Utilice el *Detective* para seguir los precedentes mencionados en una fórmula de una celda. Al rastrear estos precedentes, puede encontrar el origen de los errores. Coloque el cursor en la

celda en cuestión y luego seleccione **Herramientas > Detective > Rastrear precedentes** en el menú o pulse *Mayús+F9*. La figura 36 muestra un ejemplo sencillo de rastreo de precedentes.

Esto nos permite verificar las celdas o intervalos de origen en busca de errores que nos hayan hecho consultar el resultado del cálculo. Si el origen es un intervalo de celdas se resaltará en azul.

En otros casos, es posible rastrear un error. Para esto, utilice la función que se encuentra en **Herramientas > Detective > Rastrear error**, con la que se pueden encontrar las celdas que causan el error.

Figura 36: Trazar precedentes usando el Detective

	A	B
1	Capital	200.000,00 €
2	Años	1
3	Interés	7,50 %
4	Rentabilidad anual	15.000,00 €
5		

Para más información, busque «Detective» en el sistema de ayuda.

Ejemplos de funciones

Para los principiantes, las funciones son una de las características más intimidantes de LibreOffice Calc. Los usuarios nóveles comprenden rápidamente que las funciones son una característica importante de las hojas de cálculo, pero hay cientos de ellas y muchas requieren información que suponen conocimientos especializados. Afortunadamente, Calc incluye docenas de funciones que cualquiera puede usar, como se detalla a continuación:

Aritmética y estadística básica

Las funciones más básicas crean fórmulas para aritmética básica o para evaluar números en un intervalo de celdas.

Aritmética básica

Las funciones aritméticas básicas son: suma, resta, multiplicación y división. A excepción de la resta, cada una de estas operaciones tiene su propia función:

- **SUMA** para la suma
- **PRODUCTO** para la multiplicación
- **COCIENTE** para la división

Las funciones **SUMA**, **PRODUCTO** y **COCIENTE** son útiles para ingresar intervalos de celdas de la misma manera que cualquier otra función, con argumentos entre paréntesis tras su nombre.

Para las ecuaciones básicas, muchos usuarios prefieren los símbolos tradicionales para estas operaciones, utilizando el signo más (+) para la suma, el guion (-) para la resta, el asterisco (*) para la multiplicación y barra (/) para la división. Estos símbolos son de entrada rápida, sin necesidad de que sus manos se aparten del teclado. También está disponible una opción similar para la potencia. En lugar de ingresar = **POTENCIA**(A1,2), puede ingresar = **A1^2**.

Además, los símbolos tradicionales tienen la ventaja de que es posible ingresar fórmulas con ellos en un orden que se aproxima un poco más al formato legible por humanos, en comparación con el formato legible por la hoja de cálculo (la función equivalente). Por ejemplo, en lugar de ingresar = **SUMA**(A1:A2) o también =**SUMA**(A1,A2), escriba = **A1+A2**. Este formato,

legible e intuitivo para los humanos, es especialmente útil para operaciones compuestas, donde escribir $= A1*(A2+A3)$ es más breve y más fácil de leer que $= PRODUCTO(A1, SUMA(A2:A3))$.

La principal desventaja de usar operadores aritméticos es que no puede usar directamente un intervalo de celdas. En otras palabras, para ingresar el equivalente de $= SUMA(A1:A3)$, deberá escribir $= A1+A2+A3$.

El uso de una función o un operador depende en gran medida del usuario, excepto, por supuesto, cuando se trata de restar. Sin embargo, si usa hojas de cálculo con regularidad en un entorno grupal, como una clase o una oficina, es posible que desee estandarizar un formato de entrada para que todos los que manejan una hoja de cálculo se acostumbren a una entrada estándar.

Estadísticas simples

Otro uso común de las funciones de la hoja de cálculo es extraer información útil de una lista, como una serie de puntuaciones de pruebas en una clase o un resumen de las ganancias por trimestre de una empresa.

Por supuesto, puede analizar una lista de cifras si desea información básica, como el valor más alto, más bajo o el promedio. El único problema es que cuanto más larga sea la lista, más tiempo perderá y es más probable que se pierda en lo que está buscando. La alternativa más rápida y eficiente es usar una función. Tales razones explican la existencia de una función como **CONTAR**, que devuelve el número total de entradas en el intervalo de celdas indicado.

De manera similar, para encontrar el valor más alto o más bajo, puede usar **MIN** o **MAX**. Para cada una de estas fórmulas, todos los argumentos son un intervalo de celdas o una serie de celdas separadas por punto y coma.

Estas dos funciones tienen otra función relacionada: **MINA** o **MAXA** respectivamente, que realiza la misma función pero también trata una celda cuyo contenido sea texto como si tuviera un valor de 0 (cero). El mismo tratamiento del texto se realiza en cualquier variante de otra función que termine con "A". Cualquiera de las dos funciones da el mismo resultado y podría ser útil si utiliza una notación de texto para indicar, por ejemplo, si algún estudiante estuvo ausente cuando se redactó una prueba y desea programar un examen de recuperación.

Para mayor flexibilidad en operaciones similares, puede usar **K.ESIMO.MAYOR** o **K.ESIMO.MENOR**, las cuales agregan un argumento especializado de posición (*posición c*) para el número devuelto. Si el argumento *posición c* usado con **K.ESIMO.MAYOR** es 1, se obtiene el mismo resultado que se obtendría con **MAX**. Sin embargo, si el argumento *posición c* es 2, entonces el resultado será el segundo resultado mayor. De manera similar, un argumento *posición c* de 2 usado con **K.ESIMO.MENOR** devolverá el segundo número menor. Tanto **K.ESIMO.MAYOR** como **K.ESIMO.MENOR** son útiles como control permanente ya que al cambiar el valor del argumento *posición c* puede analizar rápidamente varios resultados.

Necesitaría ser un experto para querer encontrar la distribución de *Poisson* de una muestra o para encontrar el sesgo o binomio negativo de una distribución (y si lo es, encontrará funciones en Calc para tales cosas). Sin embargo, para un el resto de usuarios, existen funciones estadísticas más simples que se pueden aprender a usar rápidamente.

En particular, si necesita un promedio tiene varias funciones para elegir. Puede encontrar la media aritmética – es decir, el resultado de sumar todas las entradas en una lista y luego dividir el total entre el número de entradas – si ingresa un intervalo de números y usa **PROMEDIO** o si el intervalo incluirá entradas de texto, **PROMEDIOA** para darles un valor de cero.

Además, puede obtener otra información sobre un conjunto de datos:

- **MEDIANA**: Clasifica lógicamente los números (de menor a mayor) para evaluar el valor central. En un conjunto que contiene un número impar de valores, la mediana será el número de en medio de la lista clasificada. En un conjunto que contiene un número par

de valores, la mediana será la media de los dos valores de en medio de la lista clasificada.

- **MODA**: La entrada más frecuente en una lista de números.
- **CUARTIL**: La entrada en una posición establecida en la matriz de números. Además del intervalo de celdas, se especifica el tipo de cuartil: **0** para la entrada más baja, **1** para el valor del **25%**, **2** para el valor del **50%**, **3** para el **75%** y **4** para la entrada más alta. Tenga en cuenta que el resultado para los tipos 1 a 3 puede no representar un elemento real ingresado.
- **JERARQUIA**: La posición de una entrada determinada en la lista completa, medida de arriba hacia abajo o de abajo hacia arriba. Debe especificar la dirección de la celda para la entrada, el intervalo de celdas de entradas y el tipo de intervalo de celdas (**0** para el intervalo de celdas partiendo desde el más alto o **1** para el intervalo de celdas desde el más bajo).

Algunas de estas funciones se superponen; por ejemplo, los mismos resultados de **MIN** y **MAX** pueden obtenerse con **CUARTIL**. En otros casos, una clasificación o un filtro personalizados pueden dar el mismo resultado. Por dicha razón el uso de uno u otro dependerá de sus necesidades. Algunos pueden preferir usar **MIN** y **MAX** porque son fáciles de recordar, mientras que otros pueden preferir **CUARTIL** porque es más versátil.

Uso de estas funciones

En algunos casos, es posible que pueda obtener resultados similares a algunas de estas funciones configurando un filtro o una clasificación personalizada. Sin embargo, en general, las funciones se ajustan más fácilmente que los filtros o las ordenaciones de datos y brindan una gama de posibilidades más amplia.

Es posible que desee ingresar una o más fórmulas temporalmente en una celda en blanco y eliminarla una vez que haya terminado. Sin embargo, si se encuentra usando las mismas funciones constantemente, considere crear una plantilla e incluir espacio para todas las funciones que usa, con una celda a la izquierda usada como etiqueta para estas. Una vez que haya creado la plantilla, puede actualizar fácilmente cada fórmula a medida que cambien las entradas, ya sea automáticamente y sobre la marcha o pulsando la tecla **F9** para actualizar todas las celdas seleccionadas.

No importa cómo use estas funciones, probablemente las encontrará fáciles de usar y adaptables para muchos propósitos. Cuando haya dominado este conjunto, estará listo para probar funciones más complejas.

Redondeo de cifras

Para fines estadísticos y matemáticos, Calc incluye una variedad de formas para redondear números. Si es programador, también puede estar familiarizado con algunos de estos métodos. Sin embargo, no es necesario ser un especialista para encontrar utilidad en los mismos. Es posible que desee redondear por motivos de facturación o porque los lugares decimales no se traducen bien en el mundo físico; por ejemplo, si las piezas que necesita vienen en paquetes de 100 y el hecho de que solo necesita 66 es irrelevante; entonces necesita redondear para realizar el pedido. Si se familiariza con las opciones para redondear hacia arriba o hacia abajo, puede hacer que sus hojas de cálculo sean más útiles en su trabajo cotidiano.

Cuando usa una función de redondeo, tiene dos opciones para configurar sus fórmulas. Si lo desea, también puede anidar un cálculo dentro de una de las funciones de redondeo. Por ejemplo, la fórmula `=REDONDEAR((SUMA(A1, A2)))` suma las cifras en las celdas **A1** y **A2** y luego las redondea al número entero más cercano. Sin embargo, aunque no es necesario trabajar con cifras exactas todos los días, es posible que desee consultarlas ocasionalmente. Si ese es el caso, probablemente sea mejor que separe las dos funciones, para lo cual primero

escriba `= SUMA(A1,A2)` en A3 y `= REDONDEAR(A3)` en A4, para etiquetar claramente cada función.

Para más detalles sobre los métodos de redondeo, consulte la Ayuda de LibreOffice.

Funciones volátiles y no volátiles

La versión 1.3 del formato OpenDocument para documentos de oficina incluye la siguiente definición: «Las funciones que se recalculan siempre cuando hay un recálculo se denominan funciones volátiles».

Para comprender el funcionamiento de una función volátil dentro de Calc, considere el siguiente ejemplo con la función `ALEATORIO`: tras crear una hoja de cálculo en blanco e introducir la fórmula `= ALEATORIO` en la celda A1, Calc muestra un número aleatorio entre 0 y 1 en dicha celda. Si a continuación escribe cualquier valor en una celda diferente (en este ejemplo se aplicará en B2) y pulsa *Intro*, notará que el valor que se muestra en A1 se actualiza para mostrar un número aleatorio diferente. Calc regenera el número aleatorio en A1, a pesar de que el usuario no ha cambiado la fórmula en A1 y a pesar de que la fórmula de B2 actualizada no tiene ningún vínculo con A1. En resumen, la función `ALEATORIO` generará un nuevo valor cuando se actualice cualquier celda al dirigirse a **Datos > Calcular > Recalcular**, pulsando *F9* o tras cualquier entrada de datos.

Si crea una hoja de cálculo de gran tamaño, en la que el recálculo frecuente pueden afectar negativamente al rendimiento, reconocerá la importancia de conocer las funciones volátiles. Para ello asegúrese de que la hoja de cálculo en la que esté trabajando tenga el diseño adecuado para utilizar o no las funciones volátiles.

Las funciones volátiles de Calc son:

- | | |
|--------------------------------|--------------------------|
| • <code>AHORA</code> | • <code>FORMULA</code> |
| • <code>ALEATORIO</code> | • <code>INDIRECTO</code> |
| • <code>ALEATORIO.ENTRE</code> | • <code>INFO</code> |
| • <code>DESREF</code> | • <code>HOY</code> |

Calc proporciona funciones equivalentes, no volátiles, para `ALEATORIO` y `ALEATORIO.ENTRE`: estas son `ALEATORIO.NV` y `ALEATORIO.ENTRE.NV`. Pueden ser útiles cuando no se requiere que los valores de la función se actualicen con tanta frecuencia. Una función no volátil no se recalcula con nuevos eventos de entrada ni tampoco al seleccionar en **Datos > Calcular > Recalcular** o al pulsar *F9*, excepto cuando selecciona la celda que contiene la función. Las funciones no volátiles se recalculan al abrir el archivo.

Uso de comodines y expresiones regulares en funciones

Calc admite el uso de expresiones regulares y comodines en los argumentos de muchas de sus funciones.

Las expresiones regulares ofrecen un método más potente para la búsqueda de cadenas de texto. Para más información sobre las expresiones regulares, consulte la sección «Buscar y filtrar con expresiones regulares» en el «Capítulo 2, Introducir y editar datos».

Si la compatibilidad con Microsoft Excel es importante en su trabajo, tenga en cuenta que no podrá usar la potencia que brindan las expresiones regulares puesto que Excel no contiene un equivalente. En estos casos siguen estando disponibles los comodines y aunque no son tan

eficaces como las expresiones regulares, pueden exportarse al formato de Excel sin pérdida de información. Un comodín se refiere a un carácter especial que se puede sustituir uno o varios caracteres (sin especificación implícita). Los comodines hacen las búsquedas de cadenas de texto mas potentes, aunque menos específicas.

Los comodines disponibles son


- **?** (signo de interrogación) representa un caracter cualquiera.
- ***** (asterisco) representa cualquier número de caracteres.
- **~** (tilde o virgüllla) como carácter de escape para los caracteres comodín si forman parte del texto.

El uso de estos comodines es el mismo que el descrito en el diálogo *Buscar y reemplazar* del «Capítulo 2, Introducir y editar datos» Las funciones de Calc que permiten el uso de comodines y expresiones regulares son las siguientes:

- Funciones de bases de datos (BDPROMEDIO, BDCONTAR, BDCONTARA, BDEXTRAER, BDMAX, BDMIN, BDPRODUCTO, BDESVESTP, BDESVEST, BDSUMA, BDVARP, BDVAR)
- PROMEDIO.SI, PROMEDIO.SI.CONJUNTO, CONTAR.SI, CONTAR.SI.CONJUNTO, MAX.SI.CONJUNTO, MIN.SI.CONJUNTO, SUMAR.SI, SUMAR.SI.CONJUNTO
- BUSCARH, BUSCAR, BUSCARV
- COINCIDIR
- EXP.REG (excluido si están activados los comodines)
- HALLAR

Para controlar el uso de los comodines y expresiones regulares en las funciones de Calc, vaya al menú **Herramientas > Opciones > LibreOffice Calc > Calcular** (figura 37). Las 3 opciones disponibles se excluyen entre sí y son auto-explicativas:

Figura 37: Diálogo Herramientas > Opciones > LibreOffice Calc > Calcular



- **Permitir comodines en las fórmulas:** opción predefinida cuando instala Calc.
- **Permitir expresiones regulares en las fórmulas.**

- **Ningún comodín ni expresión regular en las fórmulas.**

Otra opción relacionada a tener en cuenta del mismo diálogo es la casilla de *Los criterios de búsqueda = y <> deben aplicarse a celdas enteras*, dentro del área de *Cálculos generales*. Esta controla si los criterios de búsqueda deben coincidir exactamente con toda la celda.

SUGERENCIA

Cuando están seleccionadas tanto las opciones *Los criterios de búsqueda = y <> deben aplicarse a celdas enteras* y *Permitir comodines en las fórmulas*, Calc se comporta igual que Microsoft Excel al buscar celdas en las funciones de base de datos.

De forma predeterminada, las búsquedas con expresiones regulares dentro de las funciones de Calc no distinguen entre mayúsculas y minúsculas, independientemente de que esté activada o no la casilla de *Distinguir mayúsculas y minúsculas*. Sin embargo, en algunas funciones las expresiones regulares pueden incluir la una opción «(?-i)» para cambiar a una coincidencia exacta de mayúsculas y minúsculas. Las funciones que admiten esta opción son: **BUSCAR**, **BUSCARH**, **BUSCARV**, **COINCIDIR**, **CONTAR.SI**, **CONTAR.SI.CONJUNTO**, **HALLAR**, **PROMEDIO.SI**, **PROMEDIO.SI.CONJUNTO**, **SUMAR.SI** y **SUMAR.SI.CONJUNTO**

Para ilustrar algunas de las características de las expresiones regulares, considere la hoja que se muestra en la figura 38, suponiendo que esté seleccionada la opción *Permitir expresiones regulares en las fórmulas*.

Figura 38: Uso de la función **CONTAR.SI**

A7			=	=CONTAR.SI(A\$1:A\$6;"r.d*")
	A	B	C	D
1	Fredo			
2	red			
3	Rododendro			
4	brida			
5	blusa			
6	Ridículo			
7	5			
8				

- 1) Escriba la fórmula **=CONTAR.SI(A1:A6;"r.d")** en la celda A7. Para que devuelva el valor 5 como se observa en la figura, desmarque la casilla *Los criterios de búsqueda = y <> deben aplicarse a celdas enteras* y *Distinguir mayúsculas y minúsculas*. La fórmula cuenta las celdas del intervalo que contienen r o R, cualquier carácter y d o D. Es decir: «Fredo», «red», «Rododendro», «brida» y «Ridículo».
- 2) Inserte la fórmula **=CONTAR.SI(A1:A6;"(?-i)r.d")** en la celda A7. Para que devuelva el valor 3, desmarque la casilla de *Los criterios de búsqueda = y <> deben aplicarse a celdas enteras*. Esta expresión regular también utiliza la bandera **(?-i)** para una coincidencia exacta entre mayúsculas y minúsculas. La fórmula cuenta las celdas del intervalo que contienen r, cualquier carácter y d, es decir: «Fredo», «red» y «brida».
- 3) Ingrese la fórmula **=CONTAR.SI(A1:A6;"r.d")** en la celda A7. Para que devuelva el valor 1, marque la opción *Los criterios de búsqueda = y <> deben aplicarse a celdas enteras*. La fórmula cuenta solo («red»).
- 4) Escriba la fórmula **=CONTAR.SI(A1:A6;"(?-i)R.d.*")** en la celda A7. Para que devuelva el valor 2, seleccione *Los criterios de búsqueda = y <> deben aplicarse a celdas enteras*. La fórmula cuenta las celdas del intervalo que contienen R, cualquier

carácter, d y cualquier número de caracteres, es decir: «Rododendro» y «Ridículo».se vuelve a utilizar la bandera «(?-i)» para realizar una búsqueda de coincidencia exacta entre mayúsculas y minúsculas. Por otro lado, para entender el significado de la expresión «.*» vea el siguiente ejemplo.

- 5) Escriba la fórmula `=CONTAR.SI(A1:A6;"*.r.d.*")` en la celda A7. Para que devuelva 5 seleccione la opción *Los criterios de búsqueda = y <> deben aplicarse a celdas enteras*. Aunque las opciones son las mismas que en el ejemplo 3) anterior, la expresión regular «.*» permite 0 o más caracteres tanto al estar antes de la "r" como después de la "d", por lo tanto cuenta «Fredo», «red», «Rododendro», «brida» y «Ridículo».

Es necesario mencionar que las expresiones regulares no funcionan en comparaciones simples. Por ejemplo: `A1=r.d` siempre devolverá *FALSO*, incluso si las expresiones regulares están activadas. Sólo devolverá *VERDADERO* si A1 solo contiene `r.d` literalmente (r luego un punto y luego d). Si desea hacer una comparación utilizando expresiones regulares, pruebe con la función `CONTAR.SI(A1;"r.d")` y esta devolverá los valores 1 o 0, que se interpretan como *VERDADERO* o *FALSO* y la podrá emplear en una fórmula como:

`=SI(CONTAR.SI(A1;"r.d");"Enhorabuena";"Intente de nuevo")`.

Activar la opción *Permitir expresiones regulares en las fórmulas* implica que cualquier carácter especial propio de expresiones regulares (como paréntesis o interrogación) dentro de la cadena de texto, vaya precedido de una barra inversa (\) a pesar de no formar parte de la expresión regular. Las barras invertidas deberán eliminarse si posteriormente se desactiva esta opción.

El carácter de barra inversa se utiliza en expresiones regulares para que al procesar la expresión regular el carácter siguiente no se trate como carácter especial.

Funciones avanzadas

Como es común con otros programas de hojas de cálculo, Calc se puede mejorar mediante funciones definidas por el usuario o extensiones. La configuración de funciones definidas por el usuario se puede realizar mediante macros o escribiendo complementos o extensiones.

Los conceptos básicos de escritura y ejecución de macros se tratan en el «Capítulo 14, Macros de Calc» de esta guía. Las macros se pueden vincular a menús o barras de herramientas para facilitar su ejecución o se pueden almacenar en módulos en una plantilla para que las funciones estén disponibles en otros documentos. Las macros para Calc se pueden escribir en *Basic*, *BeanShell*, *JavaScript* o *Python*.